



Today's Agenda

- Review Gallery of Program Output
- Vote on which programs are of most interest
- Talk about programs of interest
- Conclusions
- Q&A

download Programs/Data:
<http://support.sas.com/rnd/papers/>



The Old Report Days

Complex 1: Demographic Report

Complex 2: Detail Report

Instructor	BI User Courses										BI Analyst Courses										BI AppDev				Other			
	SBI	LW	SBI	SBI	SBI	SBI	SBI	SBI	SBI	SBI	SBI	SBI	SBI	SBI	SBI	LW	LW	LW	LW	LW	LW	LW	LW	LW	SBI	SBI	SBI	SBI
Adams, Alfred																												
Baffin, Barb																												
Chevell, Carla																												
DuPont, Dan																												
Edgewater, Elizabeth																												
Frankhart, Frederick																												
Gowaine, Georgia																												
Herman, Harry																												
Ignatz, Irene																												
Johannsen, Jerry																												
Kimmet, Katherine																												
LaFontaine, Larry																												
Morrison, Marie																												
Nordstrom, Norma																												
Olivier, Oscar																												
Pendergast, Peter																												
Quincy, Quentin																												
Ristow, Rita																												
Steinmetz, Steve																												
Thomas, Tony																												
Now Certified	1	2	12	4	11	9	10	1	1	8	12	10	1	2	4	3	4	2	1	1								
In Process	0	2	0	0	0	2	1	3	0	1	0	1	2	1	4	3	4	2	2	2								

Complex 3: Different Order Example 1

Every region is ordered by the region's descending sales for every product.

Region	Product	Total Sales
Asia	Slipper	\$152,032
	Men's Dress	\$119,366
	Women's Dress	\$78,234
	Boot	\$62,708
	Women's Casual	\$25,837
	Men's Casual	\$11,754
	Sandal	\$8,208
	Sport Shoe	\$2,092
Asia		\$460,231
Canada	Women's Dress	\$989,350
	Slipper	\$952,751
	Men's Dress	\$920,101
	Men's Casual	\$441,903
	Women's Casual	\$410,807
	Boot	\$385,613
	Sport Shoe	\$140,389
	Sandal	\$14,798
Canada		\$4,255,712

Complex 3: Different Order Example 2

All Products by Descending Sales

Product	Total Sales
Slipper	\$1,495,523
Women's Dress	\$1,467,025
Men's Dress	\$1,465,658
Men's Casual	\$1,116,025
Women's Casual	\$656,530
Boot	\$571,896
Sport Shoe	\$168,660
Sandal	\$71,430
Grand Total	\$7,012,737

Every region is ordered the same, but the order is determined by the overall order for all products, across all regions.

Region	Product	Total Sales
Asia	Slipper	\$152,032
	Women's Dress	\$78,234
	Men's Dress	\$119,366
	Men's Casual	\$11,754
	Women's Casual	\$25,837
	Boot	\$62,708
	Sport Shoe	\$2,092
	Sandal	\$8,208
Asia		\$460,231
Canada	Slipper	\$952,751
	Women's Dress	\$989,350
	Men's Dress	\$920,101
	Men's Casual	\$441,903
	Women's Casual	\$410,807
	Boot	\$385,613
	Sport Shoe	\$140,389
	Sandal	\$14,798
Canada		\$4,255,712

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 4: Sparkline


Division	Team	Record (24 Games)		Wins	Ties	Losses
		Sparkline 1 color	Sparkline 2 colors			
Eastern	NESUGgers			15	3	6
	SAS Macros			15	2	7
Central	MWSUG Data Steppers			16	3	5
	Pharma Sluggers			14	2	8
Western	PNW Forecasters			15	3	6
	West Coast WUSSers			15	3	6

(Green=Home Game; Gray=Away Game)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS


Complex 5: 4 Graphs in a Table



SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 6: Graphs and Tabular Report Together



SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 7: Proc Tabulate with Row Highlighting

		Retail		Wholesale		Total	
		Quantity	Amount	Quantity	Amount	Quantity	Amount
		Sum	Sum	Sum	Sum	Sum	Sum
NC	L	Missing	Missing	2272.00	45440.00	2272.00	45440.00
	M	1066.00	26600.00	1066.00	21320.00	2132.00	47920.00
	S	472.00	11800.00	472.00	9440.00	944.00	21240.00
NE	L	2421.00	60525.00	2421.00	48420.00	4842.00	108945.00
	M	1825.00	45625.00	1825.00	36500.00	3650.00	82125.00
	S	623.00	15575.00	623.00	12460.00	1246.00	28035.00
SO	L	2303.00	57575.00	2303.00	46060.00	4606.00	103635.00
	M	2149.00	54725.00	2149.00	42980.00	4298.00	97705.00
	S	1254.00	31150.00	Missing	Missing	1254.00	31150.00
WE	L	2655.00	66375.00	2655.00	53100.00	5310.00	119475.00
	M	2360.00	59000.00	2360.00	47200.00	4720.00	106200.00
	S	561.00	14025.00	561.00	11220.00	1122.00	25245.00

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 8: Demographic Report with DATA _NULL_ and TABLE Template

Demographic Report Using ODS and Table Templates

	NYHA I, II, UNK (N= 742)	NYHA III (N= 158)	NYHA IV (N= 149)
Patient Age			
N	742	158	149
Mean (SD)	70.1 (11.15)	69.0 (12.07)	69.4 (11.98)
Median	73.0	71.0	73.0
Min, Max	32, 88	32, 86	32, 86
MI Categories			
No	273 (26.0%)	42 (4.0%)	52 (4.9%)
Yes	469 (44.7%)	116 (11.0%)	97 (9.2%)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Conclusion

Program Name	DATA Step Used	Procedures Used
Complex1_demog.sas	yes	MEANS, FREQ, SORT, FORMAT, REPORT
Complex2_certify.sas	yes	SORT, FORMAT, REPORT (with macro program for multiple COMPUTE blocks)
Complex3_diff_order_report.sas	no	REPORT (for summary), SQL (for join), REPORT (for final)
Complex4_win_loss.sas	yes	FORMAT, REPORT
Complex5a_smallmult_report.sas	yes	SORT, FORMAT, SQL, GCHART, REPORT

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Conclusion

Program Name	DATA Step Used	Procedures Used
Complex6a_graph_report.sas	yes	SORT, REPORT (for summary), SQL, REPORT (uses graphs created in example 5 GCHART)
Complex7_tabulate_style.sas	no	FORMAT, TABULATE
Complex8_demog_data_null.sas	yes	TABULATE (for summary), TEMPLATE (for table template)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 1: Demographic Report

JOURNAL style;
no rules; line above
and below table and
above columns.

Bold + Indent

Using PROC REPORT
With JOURNAL Style

Patient Characteristics (N= 1,049 Patients)

Patient Demographics

Age 69.80 ± 11.409 (32 - 88)

Gender

Female 473 (45.09)

Male 576 (54.91)

LVEF 32.66 ± 17.762 (5 - 65)

CAD(required)

No 283 (26.98)

Yes 766 (73.02)

MI Status

Without MI 367 (34.99)

With MI 682 (65.01)

Hypertension Status

No Hypertension 549 (52.34)

Hypertension 500 (47.66)

NYHA Functional Class

I 376 (35.84)

II 234 (22.31)

III 158 (15.06)

IV 149 (14.20)

UNK 132 (12.58)

Underlined text

Empty line

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Initial Data Set: patient_info

Patient Info

Obs	patient_id	age	Gender	cadn	cadmi	hypern	nyha	lvef
1	120101	47	Male	0	1	1	III	54
2	120102	71	Male	1	1	0	III	59
3	120103	79	Male	1	1	1	III	54
4	120104	76	Female	0	0	0	II	31
5	120105	73	Female	1	1	1	II	24
6	120106	79	Male	1	1	1	III	53
7	120107	77	Female	0	1	0	II	9
8	120108	84	Female	1	0	0	II	55
9	120109	70	Female	0	1	0	II	28
10	120110	77	Male	1	1	0	III	37
11	120111	77	Male	0	0	0	III	47
12	120112	76	Female	0	0	1	II	52
13	120113	76	Female	1	0	0	II	30
14	120114	79	Female	1	0	1	II	21
15	120115	65	Male	1	0	1	III	48
16	120116	71	Male	1	1	1	III	20
17	120117	71	Male	1	1	0	III	21
18	120118	79	Male	1	1	0	III	52
19	120119	49	Male	0	0	0	III	20
20	120120	76	Female	1	1	0	II	50

(Print of 20 observations)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

options nodate nonumber;

** Show patient data;
proc print data=complex.patient_info(obs=20);
title 'Patient Info';
run;
ods listing close;

** Create summary information using a SAS Macro program;
options nomlogic nosymbolgen;
%macro makestat(type=sum, var=, prt=no);
  %if &type = sum %then %do;
    ods output onewayfreqs=work.&var(drop=table f_&var
                                     rename=(frequency=Count));
    proc freq data=complex.patient_info ;
      tables &var / nocum ;
    run;
  %end;
  %else %if &type = means %then %do;
    ods output summary=work.&var;
    proc means data=complex.patient_info n mean std range min max ;
      var &var;
    run;
  %end;

  %if &prt = yes %then %do;
    ods listing;
    proc print data=work.&var;
    title "&var";
    run;
    ods listing close;
  %end;
%mend makestat;

%makestat(type=means, var=age, prt=no);
%makestat(type=sum, var=gender, prt=no);
%makestat(type=means, var=lvef, prt=no);
%makestat(type=sum, var=cadmi, prt=no);
%makestat(type=sum, var=cadmi, prt=no);
%makestat(type=sum, var=hypern, prt=no);
%makestat(type=sum, var=nyha, prt=no);

```

1 Define macro program to use either FREQ or MEANS to calculate needed statistics.

2 Invoke macro program to create 7 interim data sets.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

** Combine all summary information so VARINFO is both "heading";
** and categories under the heading. Then, the VALUE variable;
** is a character variable with the extra characters;
** like parens, etc, added;
data agemeans(keep=grpord roword varinfo value brkvar);
  length grpord roword 8 varinfo $35 value $200;
  retain grpord 1 brkvar 'y';
  set age;
  varinfo = 'Age';
  value = put(age_mean, 7.2)||' ±'||put(age_stddev, 8.3)||
    ' (' ||
    compress(left(put(age_min, 2))) ||
    ' - ' ||
    compress(left(put(age_max, 3))) ||
    ')';
  roword + 1;
  call symput('n',put(age_n,comma5.0));
run;

%put----> total count = &n;

data gendcnt(keep=grpord roword varinfo value brkvar);
  length grpord roword 8 varinfo $35 value $200;
  retain grpord 2 brkvar 'y';

  set gender;
  if _n_ = 1 then do;
    varinfo = 'Gender';
    value = ' ';
    roword + 1;
    output;
  end;
  varinfo = trim(gender);
  value = put(count, 3.)||' (' ||put(percent, 6.2)||')';
  roword + 1;
  output;
run;

*** Make other data sets;
*** With same variables;

```

3 Transform data from MEANS summary data to contain a character variable called VALUE which contains the concatenated statistics and another variable called VARINFO that contains the name of the variable of interest, in this case, AGE. The macro variable &N is also created.

4 Transform data from FREQ summary data, create VARINFO and VALUE variables. Other variables created for all new data sets are: GRPORD, ROWORD and BRKVAR. These will be used to control breaking and ordering in PROC REPORT.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Work.FinalData

what the final data looks like

Obs	brkvar	grpord	roword	varinfo	value
1	y	1	1	Age	69.80 ± 11.409 (32 - 88)
2	y	2	1	Gender	
3	y	2	2	Female	473 (45.09)
4	y	2	3	Male	576 (54.91)
5	y	3	1	LVEF	32.66 ± 17.762 (5 - 65)
6	y	4	1	CAD(required)	
7	y	4	2	No	283 (26.98)
8	y	4	3	Yes	766 (73.02)
9	y	5	1	MI Status	
10	y	5	2	Without MI	367 (34.99)
11	y	5	3	With MI	682 (65.01)
12	y	6	1	Hypertension Status	
13	y	6	2	No Hypertension	549 (52.34)
14	y	6	3	Hypertension	500 (47.66)
15	y	7	1	NYHA Functional Class	
16	y	7	2	I	376 (35.84)
17	y	7	3	II	234 (22.31)
18	y	7	4	III	158 (15.06)
19	y	7	5	IV	149 (14.20)
20	y	7	6	UNK	132 (12.58)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
%let style=JOURNAL;
ods listing close;
ods rtf file='complex1_demog.rtf' style=&style;
title font='Arial' h=14pt 'Using PROC REPORT';
title2 font='Arial' h=12pt bold "With &style Style";
ods escapechar='#';

proc report data=finaldata nowd noheader ls=256
  style(report)={borderwidth=6 bordercolor=black cellpadding=3
    font_size=10pt font_face=Arial}
  style(lines)={background=white foreground=black
    font_weight=bold
    font_size=14pt font_face=Arial
    protectspecialchars=off}
  style(column)={background=white foreground=black
    font_size=10pt font_face=Arial
    font_weight=medium};
column brkvar grpord roword varinfo value;
define brkvar / order noprint;
define grpord / order noprint;
define roword / order noprint;
define varinfo /display order=data;
define value / display
  style(column)={font_weight=medium just=right asis=on};
```

- 5 Use default JOURNAL style, set as a macro variable so code could be re-run with a different style.
- 6 The report order for VARINFO and VALUE come from GRPORD and ROWORD. BRKVAR will allow a break to put the total N in a header.
- 7 BRKVAR, GRPORD and ROWORD are set to NOPRINT.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

compute varinfo ;
  if roword = 1 then do;
    call define(_col_,'style',
      "style={font_weight=bold}");
  end;
  else do;
    call define(_col_,'style',
      "style={font_weight=medium leftmargin=12pt}");
  end;
endcomp;
compute after grpord/
  style={cellheight=2pt font_size=2pt};
  line ' ';
endcomp;
compute before _page_ / style={just=center font_size=12pt};
  line "\ul{Patient Characteristics}";
  line "\ul{(N= &n Patients)}";
endcomp;
compute before brkvar / style={just=left font_size=10pt};
** needed brkvar so this could be in a sep table cell;
** and justified separately from the other 2 lines;
** and underlined;
  line "\ul{Patient Demographics}";
endcomp;
run;
ods rtf close;

```

When **8** ROWORD is 1, the report row should be BOLD (AGE, GENDER, LVEF, etc). Otherwise, indent the left margin of the cell by 12 pts.

- 9** After every group, based on GRPORD, put an empty line 2 pts high.
- 10** Use RTF control strings for special header at top of the table and above the columns. PROTECTSPECIALCHARS=OFF set in PROC REPORT statement.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 1: Alternate Demographic Report

Proc Report Alternate Demographic Example

Table 1: Multivariate Predictors of Outpatient Alcohol Treatment (n=284)

Variable	N (%)	OR (95% CI)		p-value
Age				
18-25	20 (7.04)	1		
26-30	19 (6.69)	0.932	(0.435 1.996)	0.8556
31-35	25 (8.80)	2.042	(1.069 3.895)	0.0305
36-40	95 (33.45)	2.762	(1.511 5.048)	0.001
41-45	65 (22.89)	2.331	(1.248 4.356)	0.008
>45	60 (21.13)	2.483	(1.326 4.648)	0.0045
Education level				
Less than HS	28 (9.86)	1		
HS/GED	93 (32.74)	0.948	(0.544 1.650)	0.8493
Some College	114 (40.14)	0.931	(0.532 1.628)	0.8018
College	28 (9.86)	0.479	(0.243 0.946)	0.034
Graduate School	21 (7.39)	0.723	(0.347 1.503)	0.3845
Annual Pre-tax Income Level				
None	28 (9.86)	1		
<\$20,000	170 (59.86)	1.121	(0.641 1.962)	0.6879
\$20,000 - 59,999	61 (21.48)	0.708	(0.380 1.319)	0.2766
\$60,000 - 100,000	7 (2.46)	0.447	(0.166 1.207)	0.1123
>\$100,000	18 (6.34)	0.542	(0.163 1.807)	0.319
Race/Ethnicity				
White	76 (26.76)	1		
African American	16 (5.63)	1.751	(1.183 2.590)	0.0051

Uses very similar techniques to the first example. Final data set uses slightly different structure. Style is RTF with overrides to get rid of Gray headers. PROC REPORT step almost the same as first example.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS


```

%macro compblk(colvar=,bck=,hilite=no, svar=, sval=);
  compute &colvar;

  if ord le 20 then do;
    call define (_col_, 'FORMAT', '$1. ');
    %if &hilite = y %then %do;
      if &svar.sum = &sval then
        call define(_col_, 'STYLE', "STYLE={background=&bck just=c}");
      else if mod(ord,2) = 0 then
        call define (_col_, 'STYLE',
          'STYLE={background=#eeeeee just=c}');
    %end;
    %else %if &hilite=no %then %do;
      if mod(ord,2) = 0 then
        call define (_col_, 'STYLE',
          'STYLE={background=#eeeeee just=c}');
    %end;
  end;
  else if ord in (21,22) then do;
    call define (_col_, 'FORMAT', '$3. ');
    call define (_col_, 'STYLE',
      "style={background=black foreground=white just=r}");
  end;
  else if ord = 23 then do;
    call define (_col_, 'FORMAT', '$botrow. ');
    call define (_col_, 'STYLE',
      "style={background=&bck foreground=black font_weight=bold}");
  end;
endcomp;
%mend compblk;

```

1

Needed a compute block to do traffic lighting for every course variable, used a macro program for this because could send in parameters to generate the correct compute block.

SAS

```

proc format;
  value $botrow
    '1' = 'S~nB~nI~nF~nA~nS~nT'
    '2' = 'L~nW~nS~nB~nI~nO~nV~n1'
    '3' = 'S~nB~nI~nO~nV~nR'
    '4' = 'S~nB~nI~nO~nV~nR~n1'
    '5' = 'S~nB~nI~nA~nM~nO'
    '6' = 'S~nB~nI~nW~nR~nS'
    '7' = 'S~nB~nI~nI~nD~nP'
    '8' = 'S~nB~nI~nD~nA~nS~nH'
    '9' = 'S~nB~nI~nJ~nM~nP'
    '10' = 'S~nB~nI~nS~nP'
    '11' = 'S~nB~nI~nI~nM~nS'
    '12' = 'S~nB~nI~nO~nL~nA~nP'
    '13' = 'S~nB~nI~nA~nO~nL~nA~nP'
    '14' = 'L~nW~nS~nB~nI~nP~nA~nC'
    '15' = 'L~nW~nS~nB~nI~nW~nA~nC'
    '16' = 'S~nB~nI~nA~nD~nS'
    '17' = 'L~nW~nS~nB~nI~nP~nR~nT'
    '18' = 'S~nB~nI~nT~nO~nU~nR'
    '19' = 'S~nB~nI~nV~nT~nO~nU~nR'
    '20' = 'P~nX~nB~nI~nF~nT';
run;

```

2

This format is used for the bottom line of vertical headers on the final report. Since ODS ESCAPECHAR is set to '~', '~n' puts a line feed or carriage return between every letter. (Note: If this report were going to be used with reading software, this technique would have to be changed, because it is not "friendly" for reading software. This report assumes that viewing will always be done by someone comfortable with and visually able to read the vertical headers.)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

data cert_final(keep= ord name butot batot adtot trtot
  sbifast lwsbiovl sbiovr sbiovr1 sbiamo sbiwrns sbiidp
  sbidash sbijmp sbisp sbiims sbiolap sbiaolap
  lwsbipac lwsbiwac
  sbiads lwsbiprt
  sbitour sbivtour pxbift );

length sbifast lwsbiovl sbiovr sbiovr1 sbiamo sbiwrns sbiidp
  sbidash sbijmp sbisp sbiims sbiolap sbiaolap
  lwsbipac lwsbiwac
  sbiads lwsbiprt sbitour sbivtour pxbift $5;

set cert end=eof;
array totc nsbifast nlwsbiovl nsbiovr nsbiovr1 nsbiamo nsbiwrns nsbiidp
  nsbidash nsbijmp nsbisp nsbiims nsbiolap nsbiaolap nlwsbipac nlwsbiwac
  nsbiads nlwsbiprt
  nsbitour nsbivtour npxbift;

array totp psbifast plwsbiovl psbiovr psbiovr1 psbiamo psbiwrns psbiidp
  psbidash psbijmp psbisp psbiims psbiolap psbiaolap plwsbipac nlwsbiwac
  nsbiads nlwsbiprt
  psbitour psbivtour ppxbift;

array chr $ sbifast lwsbiovl sbiovr sbiovr1 sbiamo sbiwrns sbiidp
  sbidash sbijmp sbisp sbiims sbiolap sbiaolap lwsbipac lwsbiwac
  sbiads lwsbiprt
  sbitour sbivtour pxbift;

ord = _n_;
butot = 0;
batot = 0;
adtot = 0;
trtot = 0;

```

Used arrays to hold the total of 'C' or 'P' for any course. Also used an array for the bottom row – where every course gets a number. Then can set "counter" variables on every row to see whether a set of cells should be highlighted.

```

if _n_ = 1 then do;
  do i = 1 to dim(chr);
    totc(i) = 0;
    totp(i) = 0;
  end;
end;

do i = 1 to dim(chr);
  if chr(i) = 'C' then totc(i) + 1;
  else if chr(i) = 'P' then totp(i) + 1;

  if chr(i) in ('C', 'P') then do;
    if i in (5,6,7) then do;
      butot + 1;
    end;
    else if i in (8, 9, 10,11,12,13,14,15) then do;
      batot + 1;
    end;
    else if i in (16, 17) then do;
      adtot + 1;
    end;
    else if i in (18, 19,20) then do;
      trtot + 1;
    end;
  end;
end;

output;

```

The traffic lighting was not based on 1 column value, but on whether an instructor was certified on all the courses in a group. Also, calculated BUTOT, BATOT, ADTOT and TRTOT for every row, based on "groups" col 5,6,7 is one group, etc.

```

3 if eof then do;
  ord = ord + 1;
  name='Now Certified';
  do i = 1 to dim(chr);
    chr(i) = right(put(totc(i),3.0));
  end;
  output;

  ord = ord + 1;
  name='In Process';
  do i = 1 to dim(chr);
    chr(i) = right(put(totp(i),3.0));
  end;
  output;

  ord = ord + 1;
  name='BotRow';
  do i = 1 to dim(chr);
    chr(i) = left(put(i,2.0));
  end;
  output;
end;
run;

```

- 3 At the end of the file, create 3 new rows – a total line for Certified, a total line for In Process and a line at the bottom that will use the character format defined in #2.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Final Data: work.cert_final

Obs	ord	name	butot	batot	adtot	trtot	sbifast	lwsbiovl	sbiovr	sbiovr1	sbiamo	sbiurs
1	1	Adams, Alfred	3	3	0	0				C	C	C
2	2	Baffin, Barb	3	2	1	2		C	C		C	C
3	3	Chevell, Carla	3	4	0	1			C		C	C
4	4	DuPont, Dan	3	2	0	1			C		C	C
5	5	Edgewater, Elizabeth	0	2	0	0		P				
6	6	Frankhart, Frederick	1	2	0	0						
7	7	Gawaine, Georgia	2	8	0	0	C		C		C	C
8	8	Herman, Harry	0	0	1	0						
9	9	Ignatz, Irene	3	3	1	0			C	C	C	C
10	10	Johanssen, Jerry	1	1	0	2			C	C	C	
11	11	Kimmet, Katherine	0	0	0	1		C	C	C		
12	12	LaFontaine, Larry	1	2	0	0						
13	13	Morrison, Marie	0	2	2	0			C			
14	14	Nordstrom, Norma	0	0	0	0			C			
15	15	Olivier, Oscar	3	2	2	0						P
16	16	Pendergast, Peter	3	4	0	0			C		C	C
17	17	Quincy, Quentin	3	5	0	0			C		C	C
18	18	Ristow, Rita	0	1	0	3						
19	19	Steinmetz, Steve	1	2	0	0						
20	20	Thomas, Tony	3	2	0	0		P	C		C	P
21	21	Now Certified	3	2	0	0	1	2	12	4	11	9
22	22	In Process	3	2	0	0	0	2	0	0	0	2
23	23	BotRow	3	2	0	0	1	2	3	4	5	6

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

options missing = ' ' center;

ods listing close;
ods msoffice2k file='complex2_certify.html' style=sasweb;
ods rtf file='complex2_certify.rtf';
ods escapechar='~';

proc report data=cert_final nowd
  style(report)={font_face=Arial font_size=8pt
    rules=none frame=void background=cx999999
    cellspacing=1 cellpadding=2px}
  style(column)={font_face=Arial
    font_size=8pt background=cx999999 cellwidth=.25in just=c}
  style(summary)={vjust=m}
  style(header)={vjust=b
    foreground=black background=cx999999 font_face=Arial font_size=8pt};
  column ord name butot batot adtot trtot
    ('~S={}' sbiFAST LWSBIOV1 SBIOVR SBIOVR1 )
    ('~S={background=cx999999}BI User Courses'
      SBIAMO SBIWRS SBIIDP)
    ('~S={background=orange}BI Analyst Courses'
      SBIDASH SBIJMP SBISP
      SBIIMS SBIOLAP SBIAOLAP LWSBIPAC LWSBIWAC )
    ('~S={background=cx999999}BI AppDev' SBIADS LWSBIPRT)
    ('~S={background=pink}Other' SBITOUR SBIVTOUR PXBIFT);

```

4 Report wanted for HTML or RTF. ODS ESCAPECHAR is used to reliably make vertical headers with PROC REPORT.

5 Use STYLE overrides with PROC REPORT to set general REPORT, COLUMN, SUMMARY, and HEADER defaults.

6 Spanning headers and color coding for spanning headers done here.

```

title j=c 'Certification Report';

define ord / order noprint;
define name / order 'Instructor'
  style(column)={cellwidth=1.25in just=1};
define butot /sum noprint;
define batot /sum noprint;
define adtot /sum noprint;
define trtot /sum noprint;
define sbifast / display 'S~nB~nI~nF~nA~nS~nT';
define LWSBIOV1 / display 'L~nW~nS~nB~nI~nO~nV~nI' ;
define SBIOVR / display 'S~nB~nI~nO~nV~nR' ;
define SBIOVR1 / display 'S~nB~nI~nO~nV~nR~nI' ;
define SBIAMO / display 'S~nB~nI~nA~nM~nO'
  style(header)={background=cx999999};
... Define every column for every course ...

compute name;
  if ord le 20 then do;
    if mod(ord,2) = 0 then
      call define (_col_,'STYLE', 'STYLE={background=#eeeeee just=1}');
    end;
  else if ord in (21,22) then do;
    call define (_col_,'STYLE',
      'STYLE={background=black foreground=white
        font_style=italic just=1 font_weight=bold}');
    end;
  else if ord = 23 then do;
    if name = 'BotRow' then name = ' ';
    call define (_col_,'STYLE',
      'STYLE={background=cx999999 foreground=black
        vjust=t font_size=8pt}');
    end;
endcomp;

```

7 NOPRINT the total variables.

8 Vertical headers for every course.

9 Do alternate color coding on every row for NAME and color code summary lines.

```

** doing same calc and format for every numeric cell;
** so use macro snippet above;
%compblk(colvar= sbifast, bck=cx9999cc, hilite=no);
%compblk(colvar= LWSBIOV1, bck=cx9999cc, hilite=no);
%compblk(colvar= SBIOVR , bck=cx9999cc, hilite=no);
%compblk(colvar= SBIOVR1, bck=cx9999cc, hilite=no);

%compblk(colvar= SBIAMO, bck=cx99cc99, hilite=y, svar=butot, sval=3);
%compblk(colvar= SBIWRS, bck=cx99cc99, hilite=y, svar=butot, sval=3);
%compblk(colvar= SBIIDP, bck=cx99cc99, hilite=y, svar=butot, sval=3);

%compblk(colvar= SBIDASH, bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= SBIJMP, bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= SBISP, bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= SBIIMS, bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= SBIOLAP , bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= SBIAOLAP, bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= LWSBIPAC, bck=orange, hilite=y, svar=batot, sval=8);
%compblk(colvar= LWSBIWAC, bck=orange, hilite=y, svar=batot, sval=8);

%compblk(colvar= SBIADS, bck=cxffff99, hilite=y, svar=adtot, sval=2);
%compblk(colvar= LWSBIPRT, bck=cxffff99, hilite=y, svar=adtot, sval=2);

%compblk(colvar= SBITOUR, bck=pink, hilite=y, svar=trtot, sval=3);
%compblk(colvar= SBIVTOUR, bck=pink, hilite=y, svar=trtot, sval=3);
%compblk(colvar= PXBIFT, bck=pink, hilite=y, svar=trtot, sval=3);

run;

ods _all_ close;
ods listing;

```

- 10 Call the macro program to build the COMPUTE block for every course variable. Some columns were highlighted and some are not. Macro parameters are used to control traffic lighting.

Complex 3: Different Order Example 1

Every region is ordered by the region's descending sales for every product.

1 + 2
NOPRINT variable RPORD
controls ORDER so each
region orders independently
of the other regions.

BREAK after
REGION
summary
lines

Region	Product	Total Sales
Asia	Slipper	\$152,032
	Men's Dress	\$119,366
	Women's Dress	\$78,234
	Boot	\$62,708
	Women's Casual	\$25,837
	Men's Casual	\$11,754
	Sandal	\$8,208
	Sport Shoe	\$2,092
Asia		\$460,231
Canada	Women's Dress	\$989,350
	Slipper	\$952,751
	Men's Dress	\$920,101
	Men's Casual	\$441,903
	Women's Casual	\$410,807
	Boot	\$385,613
	Sport Shoe	\$140,389
	Sandal	\$14,798
Canada		\$4,255,712

SAS GLOBAL FORUM

Input Data: sashelp.shoes

	Region	Product	Subsidiary	Number of Stores	Total Sales	Total Inventory	Total Returns
1	Africa	Sport Shoe	Luanda	2	\$801	\$3,247	\$29
2	Africa	Sport Shoe	Addis Ababa	4	\$1,690	\$16,634	\$79
3	Africa	Sport Shoe	Nairobi	10	\$2,202	\$11,328	\$174
4	Africa	Sport Shoe	Cairo	3	\$2,259	\$20,815	\$44
5	Africa	Sport Shoe	Khartoum	7	\$2,521	\$27,041	\$84
6	Africa	Sport Shoe	Algiers	9	\$2,617	\$9,372	\$168
7	Africa	Men's Dress	Cairo	5	\$4,051	\$45,962	\$97
8	Africa	Boot	Cairo	20	\$4,846	\$18,965	\$229
9	Africa	Sport Shoe	Kinshasa	10	\$4,888	\$27,998	\$162
10	Africa	Sport Shoe	Johannesburg	8	\$5,172	\$29,368	\$139
11	Africa	Boot	Luanda	8	\$6,081	\$51,572	\$325
12	Africa	Boot	Johannesburg	14	\$8,365	\$33,011	\$483
13	Africa	Women's Dress	Luanda	1	\$8,467	\$47,387	\$210
14	Africa	Men's Dress	Nairobi	1	\$8,587	\$20,877	\$363
15	Africa	Men's Casual	Khartoum	1	\$9,244	\$16,230	\$478
16	Africa	Sandal	Cairo	9	\$10,532	\$50,430	\$598
17	Africa	Sandal	Luanda	9	\$11,145	\$19,900	\$657
18	Africa	Slipper	Cairo	9	\$13,732	\$54,117	\$1,216
19	Africa	Boot	Kinshasa	16	\$13,921	\$70,736	\$553
20	Africa	Women's Dress	Cairo	3	\$14,095	\$51,145	\$745
21	Africa	Boot	Nairobi	25	\$16,282	\$66,017	\$844
22	Africa	Sandal	Nairobi	19	\$16,289	\$47,406	\$1,175
23	Africa	Sandal	Kinshasa	10	\$16,662	\$104,438	\$611
24	Africa	Sandal	Johannesburg	13	\$17,337	\$63,003	\$809
25	Africa	Women's Casual	Kinshasa	1	\$17,919	\$21,363	\$400
26	Africa	Men's Dress	Khartoum	3	\$18,053	\$51,132	\$1,177
27	Africa	Slipper	Luanda	5	\$19,146	\$97,060	\$701
28	Africa	Boot	Khartoum	24	\$19,282	\$105,370	\$700
29	Africa	Women's Casual	Khartoum	1	\$19,582	\$30,727	\$384
30	Africa	Boot	Algiers	21	\$21,297	\$73,737	\$710

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
proc report data=sashelp.shoes nowd
  out=work.prodsum(keep=region product sales
  rpord);
  where region in ('Asia', 'Canada', 'Pacific');
  column region product sales sales=rpord;
  define region / group;
  define product / group;
  define sales / sum;
  define rpord / sum 'Region Order';
run;
```

- 1 Proc REPORT can also create output data sets. In this instance, the summarized variable RPORD is used to order each region by its descending sales.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Interim Data: prodsum

	Region	Product	rpord	Sales
1	Asia	Boot	\$62,708	\$62,708
2	Asia	Men's Casual	\$11,754	\$11,754
3	Asia	Men's Dress	\$119,366	\$119,366
4	Asia	Sandal	\$8,208	\$8,208
5	Asia	Slipper	\$152,032	\$152,032
6	Asia	Sport Shoe	\$2,092	\$2,092
7	Asia	Women's Casual	\$25,837	\$25,837
8	Asia	Women's Dress	\$78,234	\$78,234
9	Canada	Boot	\$385,613	\$385,613
10	Canada	Men's Casual	\$441,903	\$441,903
11	Canada	Men's Dress	\$920,101	\$920,101
12	Canada	Sandal	\$14,798	\$14,798
13	Canada	Slipper	\$952,751	\$952,751
14	Canada	Sport Shoe	\$140,389	\$140,389
15	Canada	Women's Casual	\$410,807	\$410,807
16	Canada	Women's Dress	\$989,350	\$989,350
17	Pacific	Boot	\$123,575	\$123,575
18	Pacific	Men's Casual	\$662,368	\$662,368
19	Pacific	Men's Dress	\$426,191	\$426,191
20	Pacific	Sandal	\$48,424	\$48,424
21	Pacific	Slipper	\$390,740	\$390,740
22	Pacific	Sport Shoe	\$26,169	\$26,169
23	Pacific	Women's Casual	\$219,886	\$219,886
24	Pacific	Women's Dress	\$399,441	\$399,441

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
ods listing close;

ods html file='complex3_diff_order_by_Region.html'
      style=sasweb;

proc report data=prodsum nowd
  style(summary)=Header;
  title 'Each Region by Highest to Lowest Product Sales';
  title2 'Products Ordered Independently for Each Region';
  column region rpord product sales;
  define region /order;
  define rpord / order descending noprint; ②
  define product / order;
  define sales / sum;
  rbreak after /summarize; ③
  break after region / summarize;
  compute after;
    region = 'Grand Total';
  endcomp;
run;

ods html close;
```

② RPORD is used as an ORDER variable, but also NOPRINT.

③ Still want to do BREAK and RBREAK processing to get subtotals between each region and a grand total at the end of the report.

Complex 3: Different Order Example 2

All Products by Descending Sales

Product	Total Sales
Slipper	\$1,495,523
Women's Dress	\$1,467,025
Men's Dress	\$1,465,658
Men's Casual	\$1,116,025
Women's Casual	\$656,530
Boot	\$571,896
Sport Shoe	\$168,660
Sandal	\$71,430
Grand Total	\$7,012,737

Every region is ordered the same, but the order is determined by the overall order for all products, across all regions.

A technique similar to Example 1 was used to get the OVERALL order information into a variable called OVPORD and that variable was used to order this example.

Region	Product	Total Sales
Asia	Slipper	\$152,032
	Women's Dress	\$78,234
	Men's Dress	\$119,366
	Men's Casual	\$11,754
	Women's Casual	\$25,837
	Boot	\$62,708
	Sport Shoe	\$2,092
	Sandal	\$8,208
Asia		\$460,231
Canada	Slipper	\$952,751
	Women's Dress	\$989,350
	Men's Dress	\$920,101
	Men's Casual	\$441,903
	Women's Casual	\$410,807
	Boot	\$385,613
	Sport Shoe	\$140,389
	Sandal	\$14,798
Canada		\$4,255,712

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and SAS Global Forum are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Complex 4: Sparkline

Division	Team	Record (24 Games)		Wins	Ties	Losses
		Sparkline 1 color	Sparkline 2 colors			
Eastern	NESUGgers			15	3	6
	SAS Macros			15	2	7
Central	MWSUG Data Steppers			16	3	5
	Pharma Sluggers			14	2	8
Western	PNW Forecasters			15	3	6
	West Coast WUSSers			15	3	6

(Green=Home Game; Gray=Away Game)

1 + 4 + 6

1 + 2 + 5 + 7

3

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and SAS Global Forum are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Input Data:

```
input Division $ divord $ TeamName $ g1-g24 $;
```

```
datalines;
"Eastern",1,"SAS Macros",W,W,T,T,W,W,L,W,W,W,L,W,W,L,W,W,L,W,W,L,L
"Eastern",1,"NESUGgers",W,W,L,T,W,W,L,T,W,W,W,W,W,L,W,L,W,W,T,W,L,W
"Western",3,"West Coast WUSSers",W,L,W,W,W,T,L,W,W,W,L,W,W,L,W,T,W,T,W,W,L,W
"Western",3,"PNW Forecasters",W,W,L,W,T,W,L,T,W,W,T,W,W,L,W,W,W,L,W,W,L
"Central",2,"MWSUG Data Steppers",L,W,W,W,T,W,L,W,W,W,T,W,W,L,W,L,W,W,T,W,W,L,W,W
"Central",2,"Pharma Sluggers",W,L,W,T,L,W,L,W,W,W,L,W,W,L,W,L,W,L,W,T,W,W,L
;
run;
```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
proc format;
  value $w1 'W' = '+' ①
           'T' = '/'
           'L' = '-';
run;

%let home = ~S={font_size=30pt
                font_face='Bissantz SparkFonts 4 - Misc'
                foreground=green}; ②
%let othr = ~S={font_size=30pt
                font_face='Bissantz SparkFonts 4 - Misc'
                foreground=cx999999};

data sasteams;
  length Division $10 TeamName $20 g1-g24 $1
          sparkrec $24 divord $1
          sparkcolr $2000;

  array w1 $ g1-g24;
  infile datalines dsd;
  input Division $ divord $ TeamName $ g1-g24 $;
  sparkwin = ' ';

  wins = 0;
  losses = 0; ③
  ties = 0;
```

① Format needed for SparkFonts.

② Style override for home and away game colors.

③ Counters for team statistics.

```

do i = 1 to 24 by 1;
  sparkrec = catt(sparkrec,put(wl(i),$wl.)); ④
  if wl(i) = 'W' then wins = wins + 1;
  else if wl(i) = 'L' then losses = losses + 1;
  else if wl(i) = 'T' then ties = ties + 1;

  ** every odd-numbered game is a home game (green);
  ** every even-numbered game is field of diff team (gray);
  if mod(i,2) = 0 then do;
    sparkcolr = catt(sparkcolr,"&othr",put(wl(i),$wl.)); ⑤
  end;
  else do;
    sparkcolr = catt(sparkcolr,"&home",put(wl(i),$wl.));
  end;
end;

output;

return;
datalines;
. . . datalines . . .
run;

```

④ SPARKREC is the concatenation of all the + / - values for W, T, L.

⑤ SPARKCOLR is the same as SPARKREC, except with STYLE= overrides from macro variables set in #2.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and SAS Global Forum are product or service names or registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. It indicates SAS registration.

Report Data: work.sasteams

Work.SASTeams												
Obs	Division	TeamName		g1	g2	g3	g4	g5	g6	g7	g8	g9
1	Eastern	SAS	Macros	W	W	T	T	W	W	L	W	W
Obs	g10	g11	g12	g13	g14	g15	g16	g17	g18	g19	g20	g21
1	W	L	W	W	L	W	L	W	W	L	W	W
Obs	g22	g23	g24	4 parkrec						divord		
1	W	L	L	++//++-+++--++-+++--						1		
Obs	sparkwin	wins	losses	ties								
1		15	7	2								

Partial Print of 1 Observation in Data Set

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and SAS Global Forum are product or service names or registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. It indicates SAS registration.

Report Data: work.sasteams

[illegible]

Partial View of the SPARKCOLR variable for 1 Observation in Data Set

```
ods escapechar='~';
ods pdf file='complex4_sparkwins.pdf';
ods rtf file='complex4_sparkwins.rtf';
ods html file='complex4_sparkwins.html' style=sasweb;
proc report data=sasteams nowd;
  title 'SAS Teams Win/Loss Record';
  title2 'Displayed 2 Different Ways';
  footnotel "({~S={foreground=green}Green=Home Game;
              ~S={foreground=cx999999} Gray=Away Game~S={})";


  column divor d Division TeamName
    ('Record (24 Games)' sparkrec sparkcolr )
    wins ties losses;
  define divor d / group noprint ;
  define division / group 'Division' style(column)={vjust=m};
  define TeamName / order 'Team' style(column)={vjust=m};
  define sparkrec / display 'Sparkline 1 color'
    style(column)={font_face='Bissantz SparkFonts 4 - Misc'
                   font_size=30pt foreground=black};
  define sparkcolr / display 'Sparkline 2 colors'
    style(column)={font_size=30pt};
  define wins /sum style(column)={vjust=m} 'Wins';
  define ties/sum style(column)={vjust=m} 'Ties';
  define losses / sum style(column)={vjust=m} 'Losses';

run;
ods all close;
```

6 SPARKREC is displayed using Bissantz SparkFonts 4.

7 SPARKCOLR is also displayed using Bissantz SparkFonts 4., but through a STYLE= override.

Complex 5: 4 Graphs in a Table



SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Data: DistrictSales

	Year	District	Quarter	Sales
1	2005	Eastern	Q1	46590
2	2005	Eastern	Q1	21486
3	2005	Eastern	Q1	24733
4	2005	Eastern	Q1	28761
5	2005	Eastern	Q1	33000
6	2005	Eastern	Q1	36612
7	2005	Eastern	Q1	40056
8	2005	Eastern	Q1	42760
9	2005	Eastern	Q1	44121
10	2005	Eastern	Q1	61108
11	2005	Eastern	Q1	83212
12	2005	Eastern	Q1	106657
13	2005	Eastern	Q1	128309
14	2005	Eastern	Q1	171735
15	2005	Eastern	Q1	37271
16	2005	Eastern	Q1	38492
17	2005	Eastern	Q1	71014
18	2005	Eastern	Q1	71618
19	2005	Eastern	Q1	89416
20	2005	Eastern	Q1	151402
21	2005	Eastern	Q2	45000
22	2005	Eastern	Q2	15200
23	2005	Eastern	Q2	22810
24	2005	Eastern	Q2	51170
25	2005	Eastern	Q2	61100
26	2005	Eastern	Q2	11975
27	2005	Eastern	Q2	12211
28	2005	Eastern	Q2	15759
29	2005	Eastern	Q2	22060
30	2005	Eastern	Q2	25796

1 Summarize the data for a SAS/Graph step.

2 Use ODS and PROC GCHART to create 4 image files:
 North.png
 East1.png
 South2.png
 West3.png

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Data: diffrept1

```

3 data diffrept1(keep=rowval colval1 colval2);
  rowval = 1;
  colval1 = '1 Northern';
  colval2 = '2 Eastern';
  output;
  rowval = 2;
  colval1 = '1 Southern';
  colval2 = '2 Western';
  output;
run;

```

- 3 Create a data set with 2 rows and 3 columns. Basically this is the structure that the graphs should be displayed in.

	rowval	colval1	colval2
1	1	1 Northern	2 Eastern
2	2	1 Southern	2 Western

```

ods html path="&htout" (url=none)
      gpath="&htout" (url=none)
      file='complex5a_smallmult.html' style=sasweb stylesheet;
ods escapechar='~';

proc report data=diffrept1 nowd center missing
  style(header)={background=pink foreground=black vjust=b};
  column rowval ('Quarterly District Sales/(in Millions)'
    colval1 colval2);
  define rowval / group 'District'
    style(header)={just=l}
    style(column)={vjust=b} noprint;
  define colval1 / group ' ' order=internal;
  define colval2 / group ' ' order=internal;

```

- 4 Set the header style for the report table. Create a spanning header in the COLUMN statement.

```

compute colval1;
  colval1 = scan(colval1,1,' '); 5
  if rowval = '1' then do;
    call define('_c2_', 'STYLE', 'style={preimage="North.png" 6
      foreground=white}');
  end;
  else if rowval = '2' then do;
    call define('_c2_', 'STYLE', 'style={preimage="South2.png"
      foreground=white}');
  end;
endcomp;

compute colval2;
  colval2 = scan(colval2,1,' '); 5
  if rowval = '1' then do;
    call define('_c3_', 'STYLE', 'style={preimage="East1.png"
      foreground=white}');
  end;
  else if rowval = '2' then do;
    call define('_c3_', 'STYLE', 'style={preimage="West3.png"
      foreground=white}');
  end;
endcomp;
run;


ods _all_ close;

```

- 5 Assign a single character to COLVAL1 and COLVAL2 – to reduce white space next to the image.

- 6 Use the PREIMAGE attribute in every table cell to place the PNG files created in #2.

Complex 6: Graphs and Tabular Report Together



Report viewed in Acrobat Reader; uses graphs created in Complex 5 example.

TEXAS

```

proc report data=DistrictSales nowd
    ① out=work.ovinfo(keep=year district ovsales ovmean
        salesmn salesmx salesavg salescnt);
column district year sales sales=ovmean sales=ovsales
        sales=salesmn sales=salesmx sales=salesavg sales=salescnt;
define district / group;
define year /group;
define sales / sum;
define ovsales / sum 'Overall District Sales';
define salesmn / min 'Overall District Min';
define salesmx / max 'Overall District Max';
define salesavg/ mean 'Overall District Avg';
define salescnt / n 'Overall Number of Sales';
break after district /summarize;
rbreak after / summarize;
run;

```

- ① Summarize DistrictSales and create WORK.OVINFO for tabular portion of REPORT.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

    ②
proc format;
value $ovord 'Northern' = '1'
    'Eastern' = '2'
    'Southern' = '3'
    'Western' = '4'
    'ZZZZZZZZ' = '5';
value $onpg 'Northern' = '1'
    'Eastern' = '1'
    'Southern' = '2'
    'Western' = '2'
    'ZZZZZZZZ' = '2';
value $disting
    'Northern' = 'North.png'
    'Eastern' = 'East1.png'
    'Southern' = 'South2.png'
    'Western' = 'West3.png';
value $drtfpdf
    'Northern' = "&outpath\North.png"
    'Eastern' = "&outpath\East1.png"
    'Southern' = "&outpath\South2.png"
    'Western' = "&outpath\West3.png";
run;

```

- ② Create user-defined formats for use in step that builds final data. Need to have 2 different formats for image location – HTML is location when HTML file is VIEWED versus RTF and PDF, where the location is the image location when the RTF or PDF file is CREATED.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

options linesize=256 nocenter;
data bldmult(keep=onpg ovord district year1-year5 sale1-sale5
min1-min5 max1-max5 avgl-avg5 cnt1-cnt5
rpt_yr rpt_sl rpt_min rpt_max rpt_avg rpt_cnt);

length district $15 cr $2
rpt_yr rpt_sl rpt_min rpt_max rpt_avg rpt_cnt $250;
set ovinfo;

if year = . then year = 9999;
if district = ' ' then district = 'ZZZZZZZZ';

by district notsorted;
retain year1-year5 sale1-sale5
min1-min5 max1-max5 avgl-avg5 cnt1-cnt5
rpt_yr rpt_sl rpt_min rpt_max rpt_avg rpt_cnt;

array yr year1-year5;
array sa sale1-sale5;
array mn min1-min5;
array mx max1-max5;
array av avgl-avg5;
array cn cnt1-cnt5;

```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

if first.district then do;
i = 0;
rpt_yr = '~S={font_weight=bold background=pink }
~_Year~_
~S={~n2005~n2006~n2007~n2008~nTotal
~S={background=white}';
if district = 'ZZZZZZZZ' then do;
rpt_yr = 'All Years';
rpt_sl = ' ';
rpt_min = ' ';
rpt_max = ' ';
rpt_avg = ' ';
rpt_cnt = ' ';
end;
else if district ne 'ZZZZZZZZ' then do;
rpt_sl = '~S={font_weight=bold background=pink }
~_~_~_Total~_~_~_~S={background=white}';
rpt_min = '~S={font_weight=bold background=pink }
~_~_Min~_~_~_~S={background=white}';
rpt_max = '~S={font_weight=bold background=pink }
~_~_Max~_~_~_~S={background=white}';
rpt_avg = '~S={font_weight=bold background=pink }
~_~_Avg~_~_~_~S={background=white}';
rpt_cnt = '~S={font_weight=bold background=pink }
~_Number~_~_~_~S={background=white}';
end;
end;

```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

i + 1;

yr(i) = year;
sa(i) = ovsales;
mn(i) = salesmn;
mx(i) = salesmx;
av(i) = salesavg;
cn(i) = salescnt;

cr = '~n';
if district = 'ZZZZZZZ' then cr = ' ';

rpt_sl = compbl(catt(rpt_sl,cr,put(sa(i),dollar15.)));
rpt_min = compbl(catt(rpt_min,cr,put(mn(i),comma15.)));
rpt_max = compbl(catt(rpt_max,cr,put(mx(i),comma15.)));
rpt_avg = compbl(catt(rpt_avg,cr,put(av(i),comma15.)));
rpt_cnt = compbl(catt(rpt_cnt,cr,put(cn(i),comma15.)));

if last.district then do;
  ovord = put(district,$ovord.);
  onpg = put(district,$onpg.);
  if district = 'ZZZZZZZ' then district = 'All Districts';
  output;
end;
run;

```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Interim Data: work.bldmult

```

bldmult
Obs district      rpt_yr
1 Eastern          "S={font_weight=bold background=pink }~ Year ~S={~n2005~n2006~n2007~n2008~nTotal~S={back
2 Northern         "S={font_weight=bold background=pink }~ Year ~S={~n2005~n2006~n2007~n2008~nTotal~S={back
3 Southern         "S={font_weight=bold background=pink }~ Year ~S={~n2005~n2006~n2007~n2008~nTotal~S={back
4 Western          "S={font_weight=bold background=pink }~ Year ~S={~n2005~n2006~n2007~n2008~nTotal~S={back
5 All Districts    All Years

Obs rpt_sl
1 "S={font_weight=bold background=pink }~ Total ~S={background=white}~n $7,093,157~n $10,571,027~n $
2 "S={font_weight=bold background=pink }~ Total ~S={background=white}~n $10,736,353~n $16,048,301~n
3 "S={font_weight=bold background=pink }~ Total ~S={background=white}~n $5,935,608~n $9,237,633~n $1
4 "S={font_weight=bold background=pink }~ Total ~S={background=white}~n $10,819,902~n $16,756,694~n
5 $283,903,829

Obs rpt_min
1 "S={font_weight=bold background=pink }~ Min ~S={background=white}~n 10,930~n 14,112~n 15,794~n 27,788~n 10
2 "S={font_weight=bold background=pink }~ Min ~S={background=white}~n 11,550~n 14,687~n 17,093~n 24,570~n 11
3 "S={font_weight=bold background=pink }~ Min ~S={background=white}~n 11,753~n 12,930~n 21,078~n 38,251~n 11
4 "S={font_weight=bold background=pink }~ Min ~S={background=white}~n 10,532~n 11,676~n 13,457~n 17,443~n 10
5 10,532

Obs rpt_avg
1 "S={font_weight=bold background=pink }~ Avg ~S={background=white}~n 69,338~n 104,664~n 155,836~n 244,119~n
2 "S={font_weight=bold background=pink }~ Avg ~S={background=white}~n 120,640~n 180,325~n 274,728~n 403,329~n
3 "S={font_weight=bold background=pink }~ Avg ~S={background=white}~n 72,945~n 115,470~n 175,181~n 265,882~n
4 "S={font_weight=bold background=pink }~ Avg ~S={background=white}~n 103,047~n 159,588~n 230,816~n 348,864~n
5 189,269

Obs sale1 sale2 sale3 sale4 sale5 min1 min2 min3 min4 min5 max1
1 7003157.00 10571026.78 15739483.00 24656068.05 57969734.83 10930 14112.25 15793.99 27787.75 10930 502636.00
2 10736953.00 16048900.96 24450800.54 35896238.91 87132893.42 11550 14686.93 17092.81 24570.35 11550 757798.00
3 5835606.00 9237632.84 14014505.97 21270559.64 50358304.45 11759 12929.99 21077.99 38250.97 11759 576112.00
4 10819902.00 16756694.04 24235628.39 36630672.34 88442896.77 10532 11675.90 13457.44 17443.22 10532 1298717.00
5 283903829.46 16756694.04 24235628.39 36630672.34 88442896.77 10532 11675.90 13457.44 17443.22 10532 3431515.31

```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
ods html path="&htout" (url=none)
      file="complex6a_graf_report.html"
      style=sasweb ;
ods escapechar='~';

proc report data=bldmult nowd
  style(report)={background=white}
  style(header)={foreground=black background=pink}
  style(column)={just=right};
  title 'Proc Report with SAS/Graph Image in Table';
  column onpg ovord ('Quarterly Comparison' district)
    ('Sales' rpt_yr rpt_sl rpt_cnt rpt_min rpt_max rpt_avg);
  define onpg /order noprint;
  define ovord /order noprint;
  define district / order
    style(column)={just=left} ' ';
  define rpt_yr / display ' ';
  define rpt_sl /display ' ';
  define rpt_cnt /display ' ';
  define rpt_min /display ' ';
  define rpt_max /display ' ';
  define rpt_avg /display ' ';
```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
compute district;
  if ovord le '4' then district = ovord;
  if ovord = '1' then do;
    call define(_col_,'STYLE',
      'style={preimage="North.png" foreground=white}');
  end;
  else if ovord = '2' then do;
    call define(_col_,'STYLE',
      'style={preimage="East1.png" foreground=white}');
  end;
  else if ovord = '3' then do;
    call define(_col_,'STYLE',
      'style={preimage="South2.png" foreground=white}');
  end;
  else if ovord = '4' then do;
    call define(_col_,'STYLE',
      'style={preimage="West3.png" foreground=white}');
  end;
  else if ovord = '5' then do;
    call define(_row_,'STYLE',
      'style=Header{background=pink foreground=black}');
  end;

endcomp;
break after onpg / page;
run;
ods _all_ close;
```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Complex 7: Proc Tabulate with Row Highlighting

1 + 3 Highlight Class Levels here.

4 Highlight CITYSIZE Class Levels.

2 BACKGROUND set in CLASSLEV and VAR statements

5 Use <PARENT> to inherit style from CITYSIZE and REGION for summary cells.

		Retail		Wholesale		Total	
		Quantity	Amount	Quantity	Amount	Quantity	Amount
		Sum	Sum	Sum	Sum	Sum	Sum
NC	L	Missing	Missing	2272.00	45440.00	2272.00	45440.00
	M	1066.00	26600.00	1066.00	21320.00	2132.00	47920.00
	S	472.00	11800.00	472.00	9440.00	944.00	21240.00
NE	L	2421.00	60525.00	2421.00	48420.00	4842.00	108945.00
	M	1825.00	45625.00	1825.00	36500.00	3650.00	82125.00
	S	623.00	15575.00	623.00	12460.00	1246.00	28035.00
SO	L	2303.00	57575.00	2303.00	46060.00	4606.00	103635.00
	M	2149.00	54725.00	2149.00	42980.00	4298.00	97705.00
	S	1254.00	31150.00	Missing	Missing	1254.00	31150.00
WE	L	2655.00	66375.00	2655.00	53100.00	5310.00	119475.00
	M	2360.00	59000.00	2360.00	47200.00	4720.00	106200.00
	S	561.00	14025.00	561.00	11220.00	1122.00	25245.00

Understanding <Parent>

		Retail		Wholesale		Total	
		Quantity	Amount	Quantity	Amount	Quantity	Amount
		Sum	Sum	Sum	Sum	Sum	Sum
NC	L	Missing	Missing	2272.00	45440.00	2272.00	45440.00
	M	1066.00	26600.00	1066.00	21320.00	2132.00	47920.00
	S	472.00	11800.00	472.00	9440.00	944.00	21240.00
NE	L	2421.00	60525.00	2421.00	48420.00	4842.00	108945.00
	M	1825.00	45625.00	1825.00	36500.00	3650.00	82125.00
	S	623.00	15575.00	623.00	12460.00	1246.00	28035.00

- D. CITYSIZE CLASSLEV parent is REGION;
E. REGION CLASSLEV color is set with traffic lighting (\$REGCOL.)

- A. Header is the parent for the SALETYPE CLASSLEV values and the ALL keyword statistic;
B. QUANTITY and AMOUNT parents are the headers above each variable;
C. SUM keyword statistic parents are the variable headers above each SUM column.

In the ROW dimension, nested headings operate from left to right. So the parent cells of the rightmost heading cells are the CLASSLEV cells immediately to the left or else the ROWHEADER style sets the default inheritance (not the CLASS heading attributes).

Input Data: work.citypop

	region	citysize	pop	product	salestype	quantity	amount
1	NC	S	25000	A100	R	150	3750
2	NE	S	37000	A100	R	200	5000
3	SO	S	48000	A100	R	410	10250
4	WE	S	32000	A100	R	180	4500
5	NC	M	125000	A100	R	350	8750
6	NE	M	237000	A100	R	600	15000
7	SO	M	348000	A100	R	710	17750
8	WE	M	432000	A100	R	780	19500
9	NE	L	837000	A100	R	800	20000
10	SO	L	748000	A100	R	760	19000
11	WE	L	932000	A100	R	880	22000
12	NC	S	25000	A100	W	150	3000
13	NE	S	37000	A100	W	200	4000
14	WE	S	32000	A100	W	180	3600
15	NC	M	125000	A100	W	350	7000
16	NE	M	237000	A100	W	600	12000
17	SO	M	348000	A100	W	710	14200
18	WE	M	432000	A100	W	780	15600
19	NC	L	625000	A100	W	750	15000
20	NE	L	837000	A100	W	800	16000
21	SO	L	748000	A100	W	760	15200
22	WE	L	932000	A100	W	880	17600
23	NC	S	25000	A200	R	165	4125
24	NE	S	37000	A200	R	215	5375
25	SO	S	48000	A200	R	425	10425

SAS GLOBAL FORUM

(View Partial Rows, All Columns)

2008 SAN ANTONIO TEXAS

```
proc format;
  value $salefmt 'R'='Retail'
                'W'='Wholesale';

  value $regcol  'NC'='beige'
                'NE'='pink'
                'SO'='beige'
                'WE'='pink';
run;
```

- 1 Define region background color in format.

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
ods html body="complex7_tabulate_style.html" style=egdefault;

proc tabulate data=citypop;
  class region citysize saletype;
  classlev saletype / s={background=pink foreground=black};
  classlev region / s={background=$regcol. foreground=black};
  classlev citysize / s=<parent>{foreground=black};
  var quantity amount / s={background=pink foreground=black};

  table region*citysize*{s=<parent>{foreground=black}},
    (saletype=' ' all)*(quantity amount) /
    indent=0
    box={s={background=pink foreground=black}}
    s={background=black}
    misstext="Missing";
  keyword sum all / s={background=pink foreground=black};
  format saletype $salefmt.;
  label quantity="Quantity" amount="Amount"
    region = "Reg" citysize="Size";
  /*- Label for Reg and Size not used because of indent=0 -*/
  keylabel all="Total" ;
run;
ods html close;
```

- 2 Code background=pink for SALESTYPE (and QUANTITY and AMOUNT).
- 3 Use the \$REGCOL format for the Class levels of REGION.
- 4 Specify <PARENT> for the CITYSIZE variable Class levels.
- 5 Specify <PARENT> for the crossing of REGION*CITYSIZE – which causes the calculated statistics to take on the parent style attributes.

Complex 8: Demographic Report with DATA _NULL_ and TABLE Template

Demographic Report Using ODS and Table Templates

	NYHA I, II, UNK (N= 742)	NYHA III (N= 158)	NYHA IV (N= 149)
Patient Age			
N	742	158	149
Mean (SD)	70.1 (11.15)	69.0 (12.07)	69.4 (11.98)
Median	73.0	71.0	73.0
Min, Max	32, 88	32, 86	32, 86
MI Categories			
No	273 (26.0%)	42 (4.0%)	52 (4.9%)
Yes	469 (44.7%)	116 (11.0%)	97 (9.2%)

2 Row Identifier bold

3 Level values indented

1 Count in Header

4 Customized statistics in cells

Input Data

Input file is summarized:

	NYHA_Type	Age	Myocardial Infarction Category	Left Ventricular Ejection Fraction
1	III	47	1	54
2	III	71	1	59
3	III	79	1	54
4	I, II, UNK	76	0	31
5	I, II, UNK	73	1	24
6	III	79	1	53
7	I, II, UNK	77	1	9
8	I, II, UNK	84	0	55
9	I, II, UNK	70	1	28
10	III	77	1	37
11	III	77	0	47
12	I, II, UNK	76	0	52
13	I, II, UNK	76	0	30
14	I, II, UNK	79	0	21
15	III	65	0	48
16	III	71	1	20
17	III	71	1	21

work.ptsummary

	NYHA_Type	Myocardial Infarction Category	Type of Observation	age_N	age_Mean	age_Std	age_Min	age_Max	age_Median	age_PctN_00
1	I, II, UNK		0 11	273	70.157509158	11.180706055	32	88	74	26.02478551
2	III		0 11	42	68.952380952	12.697139114	34	79	72	4.0038131554
3	IV		0 11	52	68.711538462	12.407045023	32	96	73	4.9571020019
4	I, II, UNK		1 11	469	69.937867804	11.141763071	32	96	73	44.709246302
5	III		1 11	116	68.39137931	11.886244606	32	86	71	11.05615062
6	IV		1 11	97	69.701030928	11.803217207	32	86	73	9.2469018112
7	I, II, UNK		10	742	70.056603774	11.148828944	32	88	73	70.734032412
8	III		10	158	68.981012658	12.067470365	32	86	71	15.061963775
9	IV		10	149	69.355704638	11.984829276	32	86	73	14.204003813

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Create Macro Variables

```
data _null_;
  set ptsummary;
  if _type_='10' then do;
    if NYHA_Type='I, II, UNK' then
      call symput('t1',
        '/NYHA I, II, UNK/(N=| |put(Age_N,comma4.)| |)');
    else if NYHA_Type='III' then
      call symput('t2',"/NYHA III/(N=| |put(Age_N,comma4.)| |)');
    else if NYHA_Type='IV' then
      call symput('t3',"/NYHA IV/(N=| |put(Age_N,comma4.)| |)');
  end;
run;

%let t1 = /~S={font_weight=bold font_style=roman}&t1;
%let t2 = /~S={font_weight=bold font_style=roman}&t2;
%let t3 = /~S={font_weight=bold font_style=roman}&t3;
%put &t1;
%put &t2;
%put &t3;
```

Table Template Code

```
ods path work.testtemp(update) sasuser.templat(update)
    sashelp.tmplmst(read);

proc template;
  define table demog;
    mvar t1 t2 t3;
    column RowLabel col1 col2 col3;
    define RowLabel;
      header=' ';
      cellstyle _val_ = 'Patient Age' as
        RowHeader{font_weight=bold},
        _val_ = 'MI Categories' as
        RowHeader{font_weight=bold},
        1 as Data{leftmargin=12pt};
    end;
    define col1;
      header=t1 just=C;
    end;
    define col2;
      header=t2 just=C;
    end;
    define col3;
      header=t3 just=C;
    end;
  end;
run;
```

1

2

3

SAS

2008 SAN ANTONIO TEXAS

Custom Table Design

Demographic Report Using ODS and Table Templates

	NYHA I, II, UNK (N= 742)	NYHA III (N= 158)	NYHA IV (N= 149)
#1 →			
#2 →	Patient Age		
#3 →	N	742	158
#4 →	Mean (SD)	70.1 (11.15)	69.0 (12.07)
#5 →	Median	73.0	71.0
#6 →	Min, Max	32, 88	32, 86
#7 →			
#8 →	MI Categories		
#9 →	No	273 (26.0%)	42 (4.0%)
#10 →	Yes	469 (44.7%)	116 (11.0%)

RowLabel
@1

Col1
@2

Col2
@3

Col3
@4

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Write to Table Template

```
ods rtf file='complex8_demog.rtf' style=journal;

ods escapechar='~';
data _null_;
  length RowLabel $20 col1 col2 col3 $ 120;
  title 'Demographic Report Using ODS and Table Templates';
  RowLabel=' ';
  col1=' ';
  col2=' ';
  col3=' ';
  set ptsummary;
  Age_M=left(trim(put(Age_Median,12.1)));
  Age_Num=left(trim(put(Age_N,12.)));
  file print ods=(template='demog') n=ps;
```

FILE PRINT ODS points to custom table template. N=PS means that the program will address the whole table, writing in cells as the appropriate data value is encountered (such as _TYPE_ = '10' or _TYPE_ = '11').

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```
if _type_='10' then do;
  MeanSD=left(trim(put(Age_Mean,4.1)))||
    ' ( '||trim(put(Age_std,5.2))||' )';
  MinMax=left(trim(put(Age_Min,3.)))||
    ' , '||trim(put(Age_Max,3.));
  put #2 @1 'Patient Age' /
    @1 'N' /
    @1 'Mean (SD)' /
    @1 'Median' /
    @1 'Min, Max' /;
  if NYHA_Type='I, II, UNK' then
    put #3 @2 Age_Num $12.-C /
      @2 MeanSD $12.-C /
      @2 Age_M $12.-C /
      @2 MinMax $12.-C;
  else if NYHA_Type='III' then
    put #3 @3 Age_Num $12.-C /
      @3 MeanSD $12.-C /
      @3 Age_M $12.-C /
      @3 MinMax $12.-C;
  else if NYHA_Type='IV' then
    put #3 @4 Age_Num $12.-C /
      @4 MeanSD $12.-C /
      @4 Age_M $12.-C /
      @4 MinMax $12.-C //
      @1 'MI Categories' ;
end;
```

4

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Write to Table Template

```

if _type_='10' then do;
  . . . more code . . .
  put #2 @1 'Patient Age' /
    @1 'N' /
    @1 'Mean (SD)' /
    @1 'Median' /
    @1 'Min, Max';;
  if NYHA_Type='I, II, UNK' then
    put #3 @2 Age_Num $12.-C /
      @2 MeanSD $12.-C /
      @2 Age_M $12.-C /
      @2 MinMax $12.-C;
  else if NYHA_Type='III' then
    put #3 @3 Age_Num $12.-C /
      @3 MeanSD $12.-C /
      @3 Age_M $12.-C /
      @3 MinMax $12.-C;
  else if NYHA_Type='IV' then
    put #3 @4 Age_Num $12.-C /
      @4 MeanSD $12.-C /
      @4 Age_M $12.-C /
      @4 MinMax $12.-C //
      @1 'MI Categories' ;
end;

```

Demographic Report Using ODS and Table Templates

	NYHA I, II, UNK (N= 742)	NYHA III (N= 158)	NYHA IV (N= 149)
#1			
#2 Patient Age			
#3 N	742	158	149
#4 Mean (SD)	70.1 (11.15)	69.0 (12.07)	69.4 (11.98)
#5 Median	73.0	71.0	73.0
#6 Min, Max	32, 88	32, 86	32, 86
#7			
#8 MI Categories			
#9 No	273 (26.0%)	42 (4.0%)	52 (4.9%)
#10 Yes	469 (44.7%)	116 (11.0%)	97 (9.2%)

RowLabel
@1

Col1
@2

Col2
@3

Col3
@4

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

```

if _type_='11' then do;
  put #9 @1 'No' /
    #10 @1 'Yes';
  4 Age_Info=right(put(Age_N,3.)||' '||left(put(Age_PctN_00,pct.)));
  if NYHA_Type='I, II, UNK' then do;
    if CADMI='0' then
      put #9 @2 Age_Info;
    else
      put #10 @2 Age_Info;
  end;
  else if NYHA_Type='III' then do;
    if CADMI='0' then
      put #9 @3 Age_Info;
    else
      put #10 @3 Age_Info;
  end;
  else if NYHA_Type='IV' then do;
    if CADMI='0' then
      put #9 @4 Age_Info;
    else
      put #10 @4 Age_Info;
  end;
end;
run;

ods rtf close;

```

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS


Conclusion

Program Name	DATA Step Used	Procedures Used
Complex1_demog.sas	yes	MEANS, FREQ, SORT, FORMAT, REPORT
Complex2_certify.sas	yes	SORT, FORMAT, REPORT (with macro program for multiple COMPUTE blocks)
Complex3_diff_order_report.sas	no	REPORT (for summary), SQL (for join), REPORT (for final)
Complex4_win_loss.sas	yes	FORMAT, REPORT
Complex5a_smallmult_report.sas	yes	SORT, FORMAT, SQL, GCHART, REPORT
Complex6a_graph_report.sas	yes	SORT, REPORT (for summary), SQL, REPORT (uses graphs created in example 5 GCHART)
Complex7_tabulate_style.sas	no	FORMAT, TABULATE
Complex8_demog_data_null.sas	yes	TABULATE (for summary), TEMPLATE (for table template)

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

Your Turn



SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS

About the Author

Author Cynthia Zender
Company SAS Institute, Inc.
Telephone (575) 522-3803

Comments & E-Mail Cynthia.Zender@sas.com

This presentation and sample programs are
available for download at:

<http://support.sas.com/rnd/papers>

SAS GLOBAL FORUM

2008 SAN ANTONIO TEXAS



2008 SAN ANTONIO TEXAS

16-19 MARCH 2008