
The Old Report Days

Complex 1: Demographic Report

Complex 2: Detail Report

Complex 3: Different Order Example 1

Every region is ordered by the region's 
descending sales for every product.

Complex 3: Different Order Example 2 

Every region is 
ordered the same, 
but the order is 
determined by the 
overall order for all 
products, across all 
regions.

Complex 4: Sparkline

Complex 5: 4 Graphs in a Table

Complex 6: Graphs and Tabular Report Together 

Complex 7: Proc Tabulate with Row Highlighting

Complex 8: Demographic Report with DATA 
_NULL_ and TABLE Template

Conclusion

SORT, FORMAT, SQL, GCHART, 
REPORT

yesComplex5a_smallmult_r
eport.sas

FORMAT, REPORTyesComplex4_win_loss.sas

REPORT (for summary), SQL (for join), 
REPORT (for final)

noComplex3_diff_order_re
port.sas

SORT, FORMAT, REPORT (with macro 
program for multiple COMPUTE 
blocks)

yesComplex2_certify.sas

MEANS, FREQ, SORT, FORMAT, 
REPORT

yesComplex1_demog.sas

Procedures
Used

DATA
Step
Used

Program
Name

Conclusion

TABULATE (for summary), TEMPLATE 
(for table template)

yesComplex8_demog_data
_null.sas

FORMAT, TABULATEnoComplex7_tabulate_sty
le.sas

SORT, REPORT (for summary), SQL, 
REPORT (uses graphs created in 
example 5 GCHART)

yesComplex6a_graph_rep
ort.sas

Procedures
Used

DATA
Step
Used

Program
Name

Complex 1: Demographic Report

10

8

9 Empty line

Underlined text

Bold + Indent

1
JOURNAL style; 
no rules; line above 
and below table and 
above columns.

Initial Data Set: patient_info

(Print of 20 observations)



Copyright © 2007, SAS Institute Inc. All rights reserved. 9

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. 

options nodate nonumber;

** Show patient data;

proc print data=complex.patient_info(obs=20);
title 'Patient Info';
run;
ods listing close;

** Create summary information using a SAS Macro program;
options nomlogic nosymbolgen;
%macro makestat(type=sum, var=, prt=no);
%if &type = sum %then %do;
ods output onewayfreqs=work.&var(drop=table f_&var

rename=(frequency=Count));
proc freq data=complex.patient_info ;
tables  &var / nocum ;

run;
%end;
%else %if &type = means %then %do;
ods output summary=work.&var;
proc means data=complex.patient_info n mean std range min max ;
var &var;

run;
%end;

%if &prt = yes %then %do;
ods listing;
proc print data=work.&var;
title "&var";

run;
ods listing close;

%end;
%mend makestat;
%makestat(type=means, var=age, prt=no);
%makestat(type=sum, var=gender, prt=no);
%makestat(type=means, var=lvef, prt=no);
%makestat(type=sum, var=cadn, prt=no);
%makestat(type=sum, var=cadmi, prt=no);
%makestat(type=sum, var=hypern, prt=no);
%makestat(type=sum, var=nyha, prt=no);

2

1

1

2

Define macro program to 
use either FREQ or 
MEANS to calculate 
needed statistics.

Invoke macro program to 
create 7 interim data sets.

**  Combine all summary information so VARINFO is both "heading";
**  and categories under the heading. Then, the VALUE variable;
**  is a character variable with the extra characters;
**  like parens, etc, added.;
data agemeans(keep=grpord roword varinfo value brkvar);
length grpord roword 8 varinfo $35 value $200;
retain grpord 1 brkvar 'y';
set age;
varinfo = 'Age';
value = put(age_mean, 7.2)||' ±'||put(age_Stddev, 8.3)||

' (' ||
compress(left(put(age_Min, 2.))) ||
' - ' || 
compress(left(put(age_Max, 3.))) ||
')';

roword + 1;
call symput('n',put(age_n,comma5.0));

run;

%put----> total count = &n;

data gendcnt(keep=grpord roword varinfo value brkvar);
length grpord roword 8 varinfo $35 value $200;
retain grpord 2 brkvar 'y';

set gender;
if _n_ = 1 then do;
varinfo = 'Gender';
value = ' ';
roword + 1;
output;

end;
varinfo = trim(gender);
value = put(count, 3.)||' ('||put(percent, 6.2)||')';
roword + 1;
output;

run;

*** Make other data sets;
*** With same variables;

4

3

Transform data from 
MEANS summary data to 
contain a character 
variable called VALUE 
which contains the 
concatenated statistics 
and another variable 
called VARINFO that 
contains the name of the 
variable of interest, in this 
case, AGE. The macro 
variable &N is also 
created.

3

4 Transform data from FREQ 
summary data, create 
VARINFO and VALUE 
variables. Other variables 
created for all new data sets 
are: GRPORD, ROWORD and 
BRKVAR. These will be used 
to control breaking and 
ordering in PROC REPORT.

Work.FinalData

%let style=JOURNAL;
ods listing close;
ods rtf file='complex1_demog.rtf' style=&style;
title font='Arial' h=14pt 'Using PROC REPORT';
title2 font='Arial' h=12pt bold "With &style Style";
ods escapechar='#';

proc report data=finaldata nowd noheader ls=256
style(report)={borderwidth=6 bordercolor=black cellpadding=3

font_size=10pt font_face=Arial}
style(lines)={background=white foreground=black 

font_weight=bold
font_size=14pt font_face=Arial
protectspecialchars=off}

style(column)={background=white foreground=black 
font_size=10pt font_face=Arial
font_weight=medium};

column brkvar grpord roword varinfo value;
define brkvar / order noprint;
define grpord / order noprint;
define roword / order noprint;
define varinfo /display order=data;
define value / display

style(column)={font_weight=medium just=right asis=on};

5

5

6

6

7

7

Use default JOURNAL style, set as a macro variable so code could be re-run with 
a different style.

The report order for VARINFO and VALUE come from GRPORD and ROWORD. 
BRKVAR will allow a break to put the total N in a header.
BRKVAR, GRPORD and ROWORD are set to NOPRINT.

compute varinfo ;
if roword = 1 then do;

call define(_col_,'style',
"style={font_weight=bold}");

end;
else do;

call define(_col_,'style',
"style={font_weight=medium leftmargin=12pt}");

end;
endcomp;
compute after grpord/

style={cellheight=2pt font_size=2pt};
line ' ';

endcomp;
compute before _page_ / style={just=center font_size=12pt};
line "\ul{Patient Characteristics}";
line "\ul{(N=  &n  Patients)}";

endcomp;
compute before brkvar / style={just=left font_size=10pt};
** needed brkvar so this could be in a sep table cell;
** and justified separately from the other 2 lines;
** and underlined;
line "\ul{Patient Demographics}";

endcomp;
run;
ods rtf close;

8

8

9

10

10

9 After every group, based on GRPORD, put an empty line 2 pts high.

Use RTF control strings for special header at top of the table and above the 
columns. PROTECTSPECIALCHARS=OFF set in PROC REPORT statement.

10

When 
ROWORD is 1, 
the report row 
should be BOLD 
(AGE, 
GENDER, 
LVEF, etc). 
Otherwise, 
indent the left 
margin of the 
cell by 12 pts.

Complex 1: Alternate Demographic Report

Uses very similar techniques to the first example. Final data set uses slightly different 
structure. Style is RTF with overrides to get rid of Gray headers. PROC REPORT step 
almost the same as first example.

Complex 2: Detail Report

2

1

6

10 +

3

8

9
Alternate 
row 
highlight

Spanning 
Headers

Vertical 
Header

Traffic 
lighting

Vertical 
Footer

Summary
Lines

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. 

Initial Data Set: cert

(View All Rows, Partial Columns)

%macro compblk(colvar=,bck=,hilite=no, svar=, sval=);
compute &colvar;

if ord le 20 then do;
call define (_col_,'FORMAT', '$1.');
%if &hilite = y %then %do;

if &svar..sum = &sval then 
call define(_col_,'STYLE',"STYLE={background=&bck just=c}");

else if mod(ord,2) = 0 then 
call define (_col_,'STYLE', 

'STYLE={background=#eeeeee just=c}');
%end;
%else %if &hilite=no %then %do;

if mod(ord,2) = 0 then 
call define (_col_,'STYLE', 

'STYLE={background=#eeeeee just=c}');
%end;

end;
else if ord in (21,22) then do;

call define (_col_,'FORMAT', '$3.');
call define (_col_,'STYLE',

"style={background=black foreground=white just=r}");
end;
else if ord = 23 then do;

call define (_col_,'FORMAT', '$botrow.');
call define (_col_,'STYLE',

"style={background=&bck foreground=black font_weight=bold}");
end;

endcomp;
%mend compblk;

1 Needed a compute block to do traffic lighting for every course variable, used a macro program 
for this because could send in parameters to generate the correct compute block.

1

proc format;
value $botrow
'1' = 'S~nB~nI~nF~nA~nS~nT'
'2' = 'L~nW~nS~nB~nI~nO~nV~n1'  
'3' = 'S~nB~nI~nO~nV~nR'  
'4' = 'S~nB~nI~nO~nV~nR~n1' 
'5' = 'S~nB~nI~nA~nM~nO'  
'6' = 'S~nB~nI~nW~nR~nS'  
'7' = 'S~nB~nI~nI~nD~nP'  
'8' = 'S~nB~nI~nD~nA~nS~nH'  
'9' = 'S~nB~nI~nJ~nM~nP'  
'10' = 'S~nB~nI~nS~nP'  
'11' = 'S~nB~nI~nI~nM~nS'  
'12' = 'S~nB~nI~nO~nL~nA~nP'  
'13' = 'S~nB~nI~nA~nO~nL~nA~nP'  
'14' = 'L~nW~nS~nB~nI~nP~nA~nC'  
'15' = 'L~nW~nS~nB~nI~nW~nA~nC'  
'16' = 'S~nB~nI~nA~nD~nS'  
'17' = 'L~nW~nS~nB~nI~nP~nR~nT'  
'18' = 'S~nB~nI~nT~nO~nU~nR'  
'19' = 'S~nB~nI~nV~nT~nO~nU~nR'  
'20' = 'P~nX~nB~nI~nF~nT';

run;

2

2

This format is used for the bottom line of vertical headers on the final report. Since ODS 
ESCAPECHAR is set to '~', '~n' puts a line feed or carriage return between every letter. 
(Note: If this report were going to be used with reading software, this technique would 
have to be changed, because it is not "friendly" for reading software. This report assumes 
that viewing will always be done by someone comfortable with and visually able to read 
the vertical headers.)

data cert_final(keep= ord name butot batot adtot trtot
sbifast lwsbiov1 sbiovr sbiovr1 sbiamo sbiwrs sbiidp
sbidash sbijmp sbisp sbiims sbiolap sbiaolap
lwsbipac lwsbiwac
sbiads lwsbiprt
sbitour sbivtour pxbift );

length sbifast lwsbiov1 sbiovr sbiovr1 sbiamo sbiwrs sbiidp
sbidash sbijmp sbisp sbiims sbiolap sbiaolap
lwsbipac lwsbiwac
sbiads lwsbiprt sbitour sbivtour pxbift $5;

set cert end=eof;
array totc nsbifast nlwsbiov1 nsbiovr nsbiovr1 nsbiamo nsbiwrs nsbiidp

nsbidash nsbijmp nsbisp nsbiims nsbiolap nsbiaolap nlwsbipac nlwsbiwac
nsbiads nlwsbiprt
nsbitour nsbivtour npxbift;

array totp psbifast plwsbiov1 psbiovr psbiovr1 psbiamo psbiwrs psbiidp
psbidash psbijmp psbisp psbiims psbiolap psbiaolap plwsbipac nlwsbiwac
nsbiads nlwsbiprt
psbitour psbivtour ppxbift;

array chr $ sbifast lwsbiov1 sbiovr sbiovr1 sbiamo sbiwrs sbiidp
sbidash sbijmp sbisp sbiims sbiolap sbiaolap lwsbipac lwsbiwac
sbiads lwsbiprt
sbitour sbivtour pxbift;

ord = _n_;
butot = 0;
batot = 0;
adtot = 0;
trtot = 0;

3

Used arrays to hold the total of 'C' or 'P' for any course. Also used an 
array for the bottom row – where every course gets a number. Then can 
set "counter" variables on every row to see whether a set of cells should 
be highlighted.

3

if _n_ = 1 then do;
do i = 1 to dim(chr);
totc(i) = 0;
totp(i) = 0;

end;
end;

do i = 1 to dim(chr);
if chr(i) = 'C' then totc(i) + 1;
else if chr(i) = 'P' then totp(i) + 1;

if chr(i) in ('C', 'P') then do;
if i in (5,6,7) then do;
butot + 1;

end;
else if i in (8, 9, 10,11,12,13,14,15) then do;
batot + 1;

end;
else if i in (16, 17) then do;
adtot + 1;

end;
else if i in (18, 19,20) then do;
trtot + 1;

end;
end;

end;

output;

3

The traffic lighting was not based on 1 column value, but on whether an instructor was 
certified on all the courses in a group. Also, calculated BUTOT, BATOT, ADTOT and 
TRTOT for every row, based on "groups" col 5,6,7 is one group, etc.

3

if eof then do;
ord = ord + 1;
name='Now Certified';
do i = 1 to dim(chr);
chr(i) = right(put(totc(i),3.0));

end;
output;

ord = ord + 1;
name='In Process';
do i = 1 to dim(chr);
chr(i) = right(put(totp(i),3.0));

end;
output;

ord = ord + 1;
name='BotRow';
do i = 1 to dim(chr);
chr(i) = left(put(i,2.0));

end;
output;

end;
run;

3

At the end of the file, create 3 new rows – a total line for Certified, a total line for In 
Process and a line at the bottom that will use the character format defined in #2.3

Final Data: work.cert_final

options missing = ' ' center;

ods listing close;
ods msoffice2k file='complex2_certify.html' style=sasweb;
ods rtf file='complex2_certify.rtf';
ods escapechar='~';

proc report data=cert_final nowd
style(report)={font_face=Arial font_size=8pt 

rules=none frame=void background=cxdddddd
cellspacing=1 cellpadding=2px}

style(column)={font_face=Arial 
font_size=8pt background=#cxcccccc cellwidth=.25in just=c}

style(summary)={vjust=m}
style(header)={vjust=b 

foreground=black background=cx9999cc font_face=Arial font_size=8pt};
column ord name butot batot adtot trtot

('~S={}' sbiFAST LWSBIOV1 SBIOVR SBIOVR1 )   
('~S={background=cx99cc99}BI User Courses' 

SBIAMO SBIWRS SBIIDP)  
('~S={background=orange}BI Analyst Courses' 

SBIDASH SBIJMP SBISP
SBIIMS SBIOLAP SBIaOLAP LWSBIPAC LWSBIWAC )   

('~S={background=cxffff99}BI AppDev' SBIADS LWSBIPRT)
('~S={background=pink}Other' SBITOUR SBIVTOUR PXBIFT); 

Report wanted for HTML or RTF. ODS ESCAPECHAR is used to reliably make vertical 
headers with PROC REPORT.4

4

5

Use STYLE overrides with PROC REPORT to set general REPORT, COLUMN, 
SUMMARY, and HEADER defaults.5

Spanning headers and color coding for spanning headers done here.6

6

title j=c 'Certification Report';

define ord / order noprint;
define name / order 'Instructor'

style(column)={cellwidth=1.25in just=l};
define butot /sum noprint ;
define batot /sum noprint;
define adtot/sum noprint;
define trtot /sum noprint;
define sbifast / display 'S~nB~nI~nF~nA~nS~nT';
define LWSBIOV1 / display 'L~nW~nS~nB~nI~nO~nV~n1' ; 
define SBIOVR / display 'S~nB~nI~nO~nV~nR'  ;
define SBIOVR1 / display 'S~nB~nI~nO~nV~nR~n1' ;
define SBIAMO / display 'S~nB~nI~nA~nM~nO'  

style(header)={background=cx99cc99};
. . . Define every column for every course . . .

compute name;
if ord le 20 then do;

if mod(ord,2) = 0 then 
call define (_col_,'STYLE', 'STYLE={background=#eeeeee just=l}');

end;
else if ord in (21,22) then do;

call define (_col_,'STYLE', 
'STYLE={background=black foreground=white 

font_style=italic just=l font_weight=bold}');
end;
else if ord = 23 then do;

if name = 'BotRow' then name = ' ';
call define (_col_,'STYLE',

'STYLE={background=cx9999cc foreground=black 
vjust=t font_size=8pt}');

end;
endcomp;

NOPRINT the total variables.

9

7

8
Vertical headers for 
every course.

Do alternate color coding on every row for 
NAME and color code summary lines.

** doing same calc and format for every numeric cell;
** so use macro snippet above;
%compblk(colvar= sbifast, bck=cx9999cc, hilite=no);
%compblk(colvar= LWSBIOV1, bck=cx9999cc, hilite=no); 
%compblk(colvar= SBIOVR , bck=cx9999cc, hilite=no); 
%compblk(colvar= SBIOVR1, bck=cx9999cc,  hilite=no); 

%compblk(colvar= SBIAMO, bck=cx99cc99, hilite=y, svar=butot, sval=3); 
%compblk(colvar= SBIWRS, bck=cx99cc99, hilite=y, svar=butot, sval=3); 
%compblk(colvar= SBIIDP, bck=cx99cc99, hilite=y, svar=butot, sval=3); 

%compblk(colvar= SBIDASH, bck=orange, hilite=y, svar=batot, sval=8); 
%compblk(colvar= SBIJMP, bck=orange, hilite=y, svar=batot, sval=8);            
%compblk(colvar= SBISP, bck=orange, hilite=y, svar=batot, sval=8); 
%compblk(colvar= SBIIMS, bck=orange, hilite=y, svar=batot, sval=8); 
%compblk(colvar= SBIOLAP , bck=orange, hilite=y, svar=batot, sval=8); 
%compblk(colvar= SBIAOLAP, bck=orange, hilite=y, svar=batot, sval=8); 
%compblk(colvar= LWSBIPAC, bck=orange, hilite=y, svar=batot, sval=8); 
%compblk(colvar= LWSBIWAC, bck=orange, hilite=y, svar=batot, sval=8);  

%compblk(colvar= SBIADS, bck=cxffff99, hilite=y, svar=adtot, sval=2); 
%compblk(colvar= LWSBIPRT, bck=cxffff99, hilite=y, svar=adtot, sval=2); 

%compblk(colvar= SBITOUR, bck=pink, hilite=y, svar=trtot, sval=3); 
%compblk(colvar= SBIVTOUR,bck=pink, hilite=y, svar=trtot, sval=3); 
%compblk(colvar= PXBIFT, bck=pink, hilite=y, svar=trtot, sval=3); 

run;

ods _all_ close;
ods listing;

10

Call the macro program to build the COMPUTE block for every course variable. Some columns were 
highlighted and some ere not. Macro parameters are used to control traffic lighting.

10

Complex 3: Different Order Example 1

Every region is ordered by the region's 
descending sales for every product.

1

3

2+
NOPRINT variable RPORD 
controls ORDER so each 
region orders independently 
of the other regions.

BREAK after 
REGION 
summary 
lines

Input Data: sashelp.shoes

proc report data=sashelp.shoes nowd
out=work.prodsum(keep=region product sales 

rpord);
where region in ('Asia', 'Canada', 'Pacific');
column region product sales sales=rpord;
define region / group;
define product / group;
define sales / sum;
define rpord / sum 'Region Order';

run;

Proc REPORT can also create output data sets. In this instance, the 
summarized variable RPORD is used to order each region by its descending 
sales.

1

1

Interim Data: prodsum

ods listing close;

ods html file='complex3_diff_order_by_Region.html'
style=sasweb;

proc report data=prodsum nowd
style(summary)=Header;
title 'Each Region by Highest to Lowest Product Sales';
title2 'Products Ordered Independently for Each Region';
column region rpord product sales;
define region /order;
define rpord / order descending noprint;
define product / order;
define sales / sum;
rbreak after /summarize;
break after region / summarize;
compute after;
region = 'Grand Total';

endcomp;
run;

ods html close;

2

3

RPORD is used as an ORDER variable, but also NOPRINT.2

Still want to do BREAK and RBREAK processing to get subtotals between 
each region and a grand total at the end of the report.

3

Complex 3: Different Order Example 2 

Every region is 
ordered the same, 
but the order is 
determined by the 
overall order for all 
products, across all 
regions.

A technique similar to Example 1 was used to get 
the OVERALL order information into a variable 
called OVPORD and that variable was used to 
order this example.

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. 

Complex 4: Sparkline

4 2 + 5
3

1 1 ++ 6+ 7+



Copyright © 2007, SAS Institute Inc. All rights reserved. 21

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. 

Input Data: 

datalines;
"Eastern",1,"SAS Macros",W,W,T,T,W,W,L,W,W,W,L,W,W,L,W,L,W,W,L,W,W,W,L,L
"Eastern",1,"NESUGgers",W,W,L,T,W,W,L,T,W,W,W,W,W,L,W,L,W,L,W,W,T,W,L,W
"Western",3,"West Coast WUSSers",W,L,W,W,W,T,L,W,W,W,L,W,W,L,W,L,W,T,W,T,W,W,L,W
"Western",3,"PNW Forecasters",W,W,L,W,T,W,L,T,W,W,T,W,W,L,W,L,W,W,W,L,W,W,W,L
"Central",2,"MWSUG Data Steppers",L,W,W,W,T,W,L,W,W,W,T,W,W,L,W,L,W,W,T,W,W,L,W,W
"Central",2,"Pharma Sluggers",W,L,W,T,L,W,L,W,W,W,L,W,W,L,W,W,L,W,L,W,T,W,W,L
;
run;

input Division $ divord $ TeamName $ g1-g24 $;

proc format;
value $wl 'W' = '+'

'T' = '/'
'L' = '-';

run;
%let home = ~S={font_size=30pt 

font_face='Bissantz SparkFonts 4 - Misc' 
foreground=green};

%let othr = ~S={font_size=30pt 
font_face='Bissantz SparkFonts 4 - Misc' 
foreground=cx999999};

data sasteams;
length Division $10 TeamName $20 g1-g24 $1 

sparkrec $24 divord $1
sparkcolr $2000;

array wl $ g1-g24;
infile datalines dsd;
input Division $ divord $ TeamName $ g1-g24 $;
sparkwin = ' ';

wins = 0;
losses = 0;
ties = 0;

Format needed for SparkFonts.1

1

Style override for home and away game colors.2

2

3

Counters for team statistics.3

do i = 1 to 24 by 1;
sparkrec = catt(sparkrec,put(wl(i),$wl.));
if wl(i) = 'W' then wins = wins + 1;
else if wl(i) = 'L' then losses = losses + 1;
else if wl(i) = 'T' then ties = ties + 1;

** every odd-numbered game is a home game (green);
** every even-numbered game is field of diff team (gray);
if mod(i,2) = 0 then do;

sparkcolr = catt(sparkcolr,"&othr",put(wl(i),$wl.));
end;
else do;

sparkcolr = catt(sparkcolr,"&home",put(wl(i),$wl.));
end;

end;

output;

return;
datalines;
. . . datalines . . .
run;

4

5

SPARKREC is the concatenation of all the + / - values for W, T, L.4

SPARKCOLR is the same as SPARKREC, except with STYLE= overrides 
from macro variables set in #2.

5

Report Data: work.sasteams

Partial Print of 1 Observation in Data Set

Work.SASTeams

Obs Division    TeamName g1   g2   g3   g4   g5   g6   g7   g8   g9

1  Eastern    SAS Macros   W    W    T    T    W    W    L    W W

Obs g10  g11  g12  g13  g14  g15  g16  g17  g18  g19  g20  g21

1   W    L    W    W    L    W    L    W    W    L    W    W

Obs g22    g23    g24            sparkrec divord

1   W      L      L     ++//++-+++-++-+-++-+++-- 1

Obs sparkwin wins    losses    ties

1               15        7        2

4

Report Data: work.sasteams

Partial View of the SPARKCOLR variable for 1 Observation in Data Set

5

ods escapechar='~';
ods pdf file='complex4_sparkwins.pdf';
ods rtf file='complex4_sparkwins.rtf';
ods html file='complex4_sparkwins.html' style=sasweb;
proc report data=sasteams nowd;

title 'SAS Teams Win/Loss Record';
title2 'Displayed 2 Different Ways';
footnote1 "(~S={foreground=green}Green=Home Game; 

~S={foreground=cx999999} Gray=Away Game~S={})";

column divord Division TeamName
('Record (24 Games)' sparkrec sparkcolr )
wins ties losses;

define divord / group noprint ;
define division / group 'Division' style(column)={vjust=m};
define TeamName / order 'Team' style(column)={vjust=m};;
define sparkrec / display 'Sparkline 1 color'

style(column)={font_face='Bissantz SparkFonts 4 - Misc'
font_size=30pt foreground=black};

define sparkcolr / display 'Sparkline 2 colors'
style(column)={font_size=30pt};

define wins /sum style(column)={vjust=m} 'Wins';
define ties/sum style(column)={vjust=m} 'Ties'; 
define losses / sum style(column)={vjust=m} 'Losses'; 

run;
ods _all_ close;

6
7

6

7

SPARKREC is displayed using Bissantz SparkFonts 4.

SPARKCOLR is also displayed using Bissantz SparkFonts 4., but through a STYLE= 
override.

Complex 5: 4 Graphs in a Table

Data: DistrictSales

1 Summarize the data for a SAS/Graph 
step.

2 Use ODS and PROC GCHART to 
create 4 image files:
North.png
East1.png
South2.png
West3.png

Data: diffrept1
data diffrept1(keep=rowval colval1 colval2);

rowval = 1;
colval1 = '1 Northern';
colval2 = '2 Eastern';
output;
rowval = 2;
colval1 = '1 Southern';
colval2 = '2 Western';
output;

run;

3

Create a data set with 2 rows and 3 columns. Basically this is the structure 
that the graphs should be displayed in.

3

ods html path="&htout" (url=none)
gpath="&htout" (url=none)
file='complex5a_smallmult.html' style=sasweb stylesheet;

ods escapechar='~';

proc report data=diffrept1 nowd center missing
style(header)={background=pink foreground=black vjust=b};
column rowval ('Quarterly District Sales/(in Millions)' 

colval1 colval2) ;
define rowval / group 'District' 

style(header)={just=l}
style(column)={vjust=b} noprint;

define colval1 / group ' ' order=internal;
define colval2 / group ' ' order=internal;

4 Set the header style for the report table. Create a spanning header in the COLUMN 
statement.

4

compute colval1;
colval1 = scan(colval1,1,' ');
if rowval = '1' then do;

call define('_c2_','STYLE','style={preimage="North.png"
foreground=white}');

end;
else if rowval = '2' then do;

call define('_c2_','STYLE','style={preimage="South2.png"
foreground=white}');

end;
endcomp;

compute colval2;
colval2 = scan(colval2,1,' ');
if rowval = '1' then do;

call define('_c3_','STYLE','style={preimage="East1.png" 
foreground=white}');

end;
else if rowval = '2' then do;

call define('_c3_','STYLE','style={preimage="West3.png"
foreground=white}');

end;
endcomp;
run;

ods _all_ close;

6 Use the PREIMAGE attribute in every table cell to place the PNG files created in #2.

6

5 Assign a single character to COLVAL1 and COLVAL2 – to reduce white space next to 
the image.

5

5

Complex 6: Graphs and Tabular Report Together 

Report viewed in Acrobat Reader; uses graphs created in Complex 5 example.

proc report data=DistrictSales nowd
out=work.ovinfo(keep=year district ovsales ovmean

salesmn salesmx salesavg salescnt);
column district year sales sales=ovmean sales=ovsales

sales=salesmn sales=salesmx sales=salesavg sales=salescnt;
define district / group;
define year /group;
define sales / sum;
define ovsales / sum 'Overall District Sales';
define salesmn / min 'Overall District Min';
define salesmx / max 'Overall District Max';
define salesavg/ mean 'Overall District Avg';
define salescnt / n 'Overall Number of Sales';
break after district /summarize;
rbreak after / summarize;

run; 

1 Summarize DistrictSales and create WORK.OVINFO for tabular portion of REPORT.

1

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. 

proc format;
value $ovord 'Northern' = '1'

'Eastern' = '2'
'Southern' = '3'
'Western' = '4'
'ZZZZZZZZ' = '5';

value $onpg 'Northern' = '1'
'Eastern' = '1'
'Southern' = '2'
'Western' = '2'
'ZZZZZZZZ' = '2';

value $distimg
'Northern' = 'North.png'
'Eastern' = 'East1.png'
'Southern' = 'South2.png'
'Western' = 'West3.png';

value $drtfpdf
'Northern' = "&outpath\North.png"
'Eastern' = "&outpath\East1.png"
'Southern' = "&outpath\South2.png"
'Western' = "&outpath\West3.png";

run;

2 Create user-defined formats for use in step that builds final data. Need to have 2 
different formats for image location – HTML is location when HTML file is VIEWED 
versus RTF and PDF, where the location is the image location when the RTF or PDF 
file is CREATED.

2

options linesize=256 nocenter;
data bldmult(keep=onpg ovord district year1-year5 sale1-sale5

min1-min5 max1-max5 avg1-avg5 cnt1-cnt5   
rpt_yr rpt_sl rpt_min rpt_max rpt_avg rpt_cnt);

length district $15 cr $2
rpt_yr rpt_sl rpt_min rpt_max rpt_avg rpt_cnt $250;

set ovinfo;

if year = . then year = 9999;
if district = ' ' then district = 'ZZZZZZZZ';

by district notsorted;
retain year1-year5 sale1-sale5

min1-min5 max1-max5 avg1-avg5 cnt1-cnt5
rpt_yr rpt_sl rpt_min rpt_max rpt_avg rpt_cnt;

array yr year1-year5;
array sa sale1-sale5;
array mn min1-min5;
array mx max1-max5;
array av avg1-avg5;
array cn cnt1-cnt5;

if first.district then do;
i = 0;
rpt_yr = '~S={font_weight=bold background=pink }

~_~_Year~_~_
~S={}~n2005~n2006~n2007~n2008~nTotal
~S={background=white}';

if district = 'ZZZZZZZZ' then do;
rpt_yr = 'All Years';
rpt_sl = ' ';
rpt_min = ' ';
rpt_max = ' ';
rpt_avg = ' ';
rpt_cnt = ' ';

end;
else if district ne 'ZZZZZZZZ' then do;
rpt_sl = '~S={font_weight=bold background=pink }

~_~_~_~_Total~_~_~_~_~S={background=white}';
rpt_min = '~S={font_weight=bold background=pink }

~_~_Min~_~_~S={background=white}';
rpt_max = '~S={font_weight=bold background=pink }

~_~_Max~_~_~S={background=white}';
rpt_avg = '~S={font_weight=bold background=pink }

~_~_Avg~_~_~S={background=white}';
rpt_cnt = '~S={font_weight=bold background=pink }

~_Number~_~S={background=white}';
end;

end;

i + 1;

yr(i) = year;
sa(i) = ovsales;
mn(i) = salesmn;
mx(i) = salesmx;
av(i) = salesavg;
cn(i) = salescnt;

cr = '~n';
if district = 'ZZZZZZZZ' then cr = ' ';

rpt_sl = compbl(catt(rpt_sl,cr,put(sa(i),dollar15.)));
rpt_min = compbl(catt(rpt_min,cr,put(mn(i),comma15.)));
rpt_max = compbl(catt(rpt_max,cr,put(mx(i),comma15.))) ;
rpt_avg = compbl(catt(rpt_avg,cr,put(av(i),comma15.)));
rpt_cnt = compbl(catt(rpt_cnt,cr,put(cn(i),comma15.)));

if last.district then do;
ovord = put(district,$ovord.);
onpg = put(district,$onpg.);
if district = 'ZZZZZZZZ' then district = 'All Districts';
output;

end;
run;

Interim Data: work.bldmult

ods html path="&htout" (url=none)
file="complex6a_graf_report.html" 

style=sasweb ;
ods escapechar='~';

proc report data=bldmult nowd
style(report)={background=white}
style(header)={foreground=black background=pink}
style(column)={just=right};
title 'Proc Report with SAS/Graph Image in Table';
column onpg ovord ('Quarterly Comparison' district) 

('Sales' rpt_yr rpt_sl rpt_cnt rpt_min rpt_max rpt_avg);
define onpg /order noprint;
define ovord /order noprint;
define district / order 

style(column)={just=left} ' ';
define rpt_yr / display ' ';
define rpt_sl /display ' ';
define rpt_cnt /display ' ';
define rpt_min /display ' ';
define rpt_max /display ' ';
define rpt_avg /display ' ';

compute district;
if ovord le '4' then district = ovord;
if ovord = '1' then do;

call define(_col_,'STYLE',
'style={preimage="North.png" foreground=white}');

end;
else if ovord = '2' then do;

call define(_col_,'STYLE',
'style={preimage="East1.png" foreground=white}');

end;
else if ovord = '3' then do;

call define(_col_,'STYLE',
'style={preimage="South2.png" foreground=white}');

end;
else if ovord = '4' then do;

call define(_col_,
'STYLE','style={preimage="West3.png" foreground=white}');

end;
else if ovord = '5' then do;

call define(_row_,
'STYLE','style=Header{background=pink foreground=black}');

end;

endcomp;
break after onpg / page;

run;
ods _all_ close;

Complex 7: Proc Tabulate with Row Highlighting

1 3+
Highlight
Class Levels
here.

4
Highlight
CITYSIZE 
Class Levels.

5

Use 
<PARENT> 
to inherit 
style from 
CITYSIZE 
and 
REGION for 
summary 
cells.

2

BACKGROUND 
set in 
CLASSLEV and 
VAR statements

Understanding <Parent>

D. CITYSIZE CLASSLEV 
parent is REGION;

E. REGION CLASSLEV 
color is set with traffic 
lighting ($REGCOL.)

A. Header is the parent for the 
SALETYPE CLASSLEV 
values and the ALL 
keyword statistic;

B. QUANTITY and AMOUNT 
parents are the headers 
above each variable;

C. SUM keyword statistic 
parents are the variable 
headers above each SUM 
column.

A
B
C

DE

In the ROW dimension, nested headings operate from left to right. So the parent cells of 
the rightmost heading cells are the CLASSLEV cells immediately to the left or else the 
ROWHEADER style sets the default inheritance (not the CLASS heading attributes). 

Input Data: work.citypop

(View Partial Rows, All Columns)

proc format;
value $salefmt 'R'='Retail'

'W'='Wholesale';

value $regcol 'NC'='beige'
'NE'='pink'
'SO'='beige'
'WE'='pink';

run;

1 Define region background color in format.

1

ods html body="complex7_tabulate_style.html" style=egdefault; 

proc tabulate data=citypop;
class region citysize saletype; 
classlev saletype / s={background=pink foreground=black};
classlev region     / s={background=$regcol. foreground=black};
classlev citysize / s=<parent>{foreground=black};
var quantity amount / s={background=pink foreground=black};

table region*citysize*{s=<parent>{foreground=black}},
(saletype=' ' all)*(quantity amount) /

indent=0 
box={s={background=pink foreground=black}}
s={background=black}
misstext="Missing";

keyword sum all     / s={background=pink foreground=black};
format saletype $salefmt.;
label quantity="Quantity" amount="Amount"

region = "Reg" citysize="Size";
/*- Label for Reg and Size not used because of indent=0 -*/
keylabel all="Total" ;
run;
ods html close;

Use the $REGCOL format for the Class levels of REGION. 
Specify <PARENT> for the CITYSIZE variable Class levels.

5

4

2

4

5

3

Specify <PARENT> for the crossing of REGION*CITYSIZE – which causes the calculated 
statistics to take on the parent style attributes.

3

Code background=pink for SALESTYPE (and QUANTITY and AMOUNT). 2

Complex 8: Demographic Report with DATA 
_NULL_ and TABLE Template

1

3

Count in 
Header

2
Row 
Identifier 
bold

Level
values
indented

4
Customized 
statistics in 
cells

Input Data
Input file is summarized:

work.ptsummary

Create Macro Variables
data _null_;

set ptsummary;
if _type_='10' then do;

if NYHA_Type='I, II, UNK' then 
call symput('t1',

'/NYHA I, II, UNK/(N='||put(Age_N,comma4.)||')');
else if NYHA_Type='III' then 

call symput('t2',"/NYHA III/(N="||put(Age_N,comma4.)||')');
else if NYHA_Type='IV' then 

call symput('t3',"/NYHA IV/(N="||put(Age_N,comma4.)||')');
end;

run;

%let t1 = /~S={font_weight=bold font_style=roman}&t1;
%let t2 = /~S={font_weight=bold font_style=roman}&t2;
%let t3 = /~S={font_weight=bold font_style=roman}&t3;
%put &t1;
%put &t2; 
%put &t3;

1



Copyright © 2007, SAS Institute Inc. All rights reserved. 35

Copyright © 2007, SAS Institute Inc. All rights reserved. SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. 

Table Template Code
ods path work.testtemp(update) sasuser.templat(update)

sashelp.tmplmst(read);

proc template;
define table demog;

mvar t1 t2 t3;
column RowLabel col1 col2 col3;
define RowLabel;

header=' ';
cellstyle _val_ = 'Patient Age' as 

RowHeader{font_weight=bold},
_val_ = 'MI Categories' as   

RowHeader{font_weight=bold},
1 as Data{leftmargin=12pt};

end;
define col1;

header=t1 just=C;
end;
define col2;

header=t2 just=C;
end;
define col3;

header=t3 just=C;
end;

end;
run;

2

31

Custom Table Design

RowLabel
@1

Col1
@2

Coll2
@3

Col3
@4

#1
#2
#3
#4
#5
#6

#9
#10

#7
#8

Write to Table Template
ods rtf file='complex8_demog.rtf' style=journal;

ods escapechar='~';
data _null_;

length RowLabel $20 col1 col2 col3 $ 120;
title 'Demographic Report Using ODS and Table Templates';
RowLabel=' ';
col1=' ';
col2=' ';
col3=' ';
set ptsummary;

Age_M=left(trim(put(Age_Median,12.1)));
Age_Num=left(trim(put(Age_N,12.)));
file print ods=(template='demog') n=ps;

FILE PRINT ODS points to custom table template. N=PS means that the program 
will address the whole table, writing in cells as the appropriate data value is 
encountered (such as _TYPE_ = '10' or _TYPE_ = '11').

if _type_='10' then do;
MeanSD=left(trim(put(Age_Mean,4.1))||

' ('||trim(put(Age_std,5.2))||')');
MinMax=left(trim(put(Age_Min,3.))||

', '||trim(put(Age_Max,3.)));
put #2 @1 'Patient Age' /

@1 'N' /
@1 'Mean (SD)' / 
@1 'Median' /
@1 'Min, Max'/;            

if NYHA_Type='I, II, UNK' then
put #3 @2 Age_Num $12.-C /

@2 MeanSD $12.-C /
@2 Age_M $12.-C /
@2 MinMax $12.-C;                

else if NYHA_Type='III' then
put #3 @3 Age_Num $12.-C/

@3 MeanSD $12.-C /
@3 Age_M $12.-C /
@3 MinMax $12.-C;                

else if NYHA_Type='IV' then
put #3 @4 Age_Num $12.-C/

@4 MeanSD $12.-C /
@4 Age_M $12.-C /
@4 MinMax $12.-C //
@1 'MI Categories' ;                

end;

4

Write to Table Template

RowLabel
@1

Col1
@2

Coll2
@3

Col3
@4

#1
#2
#3
#4
#5
#6

#9
#10

#7
#8

if _type_='10' then do;
. . . more code . . .
put #2 @1 'Patient Age' /

@1 'N' /
@1 'Mean (SD)' / 
@1 'Median' /
@1 'Min, Max'/;            

if NYHA_Type='I, II, UNK' then
put #3 @2 Age_Num $12.-C /

@2 MeanSD $12.-C /
@2 Age_M $12.-C /
@2 MinMax $12.-C;                

else if NYHA_Type='III' then
put #3 @3 Age_Num $12.-C/

@3 MeanSD $12.-C /
@3 Age_M $12.-C /
@3 MinMax $12.-C;                

else if NYHA_Type='IV' then
put #3 @4 Age_Num $12.-C/

@4 MeanSD $12.-C /
@4 Age_M $12.-C /
@4 MinMax $12.-C //
@1 'MI Categories' ;                

end;

if _type_='11' then do;
put #9 @1 'No' / 

#10 @1 'Yes';
Age_Info=right(put(Age_N,3.)||' '||left(put(Age_PctN_00,pct.)));
if NYHA_Type='I, II, UNK' then do;

if CADMI='0' then 
put #9 @2 Age_Info;

else
put #10 @2 Age_Info;

end; 
else if NYHA_Type='III' then do;

if CADMI='0' then 
put #9 @3 Age_Info;

else
put #10 @3 Age_Info;

end; 
else if NYHA_Type='IV' then do;

if CADMI='0' then 
put #9 @4 Age_Info;

else
put #10 @4 Age_Info;

end; 
end;       

run;

ods rtf close;

4

Conclusion

TABULATE (for summary), TEMPLATE (for table 
template)

yesComplex8_demog_data_null.sas

FORMAT, TABULATEnoComplex7_tabulate_style.sas

SORT, REPORT (for summary), SQL, REPORT (uses 
graphs created in example 5 GCHART)

yesComplex6a_graph_report.sas

SORT, FORMAT, SQL, GCHART, REPORTyesComplex5a_smallmult_report.sas

FORMAT, REPORTyesComplex4_win_loss.sas

REPORT (for summary), SQL (for join), REPORT (for 
final)

noComplex3_diff_order_report.sas

SORT, FORMAT, REPORT (with macro program for 
multiple COMPUTE blocks)

yesComplex2_certify.sas

MEANS, FREQ, SORT, FORMAT, REPORTyesComplex1_demog.sas

Procedures
Used

DATA
Step
Used

Program
Name

Your Turn

This presentation and sample programs are 
available for download at:

http://support.sas.com/rnd/papers

